INSTYTUT INFORMATYKI UNIWERSYTETU WROCŁAWSKIEGO

INSTITUTE OF COMPUTER SCIENCE UNIVERSITY OF WROCŁAW

ul. Przesmyckiego 20 51–151 Wrocław Poland

Stanisław Lewanowicz and Paweł Woźny

Algorithms for construction of recurrence relations for the coefficients of expansions in series of classical orthogonal polynomials

February 2001

ALGORITHMS FOR CONSTRUCTION OF RECURRENCE RELATIONS FOR THE COEFFICIENTS OF EXPANSIONS IN SERIES OF CLASSICAL ORTHOGONAL POLYNOMIALS^{(∗)}

STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

Abstract. Let fPk(x)g be any sequence of classical orthogonal polynomials, i.e., (i) very classical orthogonal polynomials, or (ii) classical orthogonal polynomials of a discrete variable, or (iii) q-classical orthogonal polynomials in the Hahn sense.

Maple implementation is given of an algorithmic approach to the problem of constructing recurrence relations for the coefficients of the Fourier series expansions with respect to the polynomials fPk(x)g.

1. Introduction

Let {P_k(x)} be any sequence of classical orthogonal polynomials, i.e., • very classical orthogonal polynomials (VCOP, in short) [8, pp. 150–152], or • classical orthogonal polynomials of a discrete variable (DCOP, in short) [34, Chapter II], or • q-classical orthogonal polynomials in the Hahn sense (qCOP, in short) [13, 16, 31, 32]. Given a function f, expansion in series f = ∑ k a_k [f]P_k (1.1)

is a matter of interest in numerical analysis, applied mathematics and mathematical physics. Important special cases are connection problem,

P̄_n(x) = ∑ k c_kP_k (n ∈ Z^+),

and (generalized) linearization problem,

P̄_m(x)P̄_n(x) = ∑ k l_kP_k (m, n ∈ Z^+),

where {P̄_k(x)} is a family of polynomials (e.g., belonging to the same class as {P_k (x)}). In particular, positivity of the connection coefficients c_k, or linearization coefficients l_k is of great importance. See [1, 2, 3, 4, 5, 10, 21, 22, 23, 27, 29, 44, 45, 46].

Usually, the determination of the expansion coefficients a_k [f] requires a deep knowledge of special functions (from hypergeometric, or basic hypergeometric families). It is important to remark that, even in the case when it is possible to compute explicitly these coefficients, it is often useful to have a recurrence relation of the type r∑ i=0 A_i(k) a_{k+i}[f] = B(k), (1.2)

where r ∈ N, and A_i and B are known functions of k. Equation (1.2) may serve as a tool for detection certain properties of a_k [f], or for numerical evaluation of these quantities.

(∗)Supported by Komitet Badań Naukowych (Poland) under the grant 2P03A02717.

Corrected and updated version. May 20, 2003.

2 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

In the present paper, we give Maple algorithms implementing the methods described in [26, 28, 48], which, under assumption that the function f(x) satisfies a linear differential equation (VCOP case), or linear difference (q-difference) equation (DCOP and qCOP cases) with polynomial coefficients, construct the recurrence relations for the coefficients a_k [f] of the form (1.2), where the functions A_i are given in terms of the polynomial coefficients σ and τ of the Pearson-type q-difference equation for the orthogonality weight ϱ associated with {P_k(x)}, and the coefficients of the three-term recurrence relation and of the structure relations satisfied by the polynomials {P_k(x)}.

2. Very classical polynomials 2.1. Properties of the very classical orthogonal polynomials. Let P^{(α,β)}_n (x), L^α_n (x), Y^α_n (x) and H_n(x) denote the monic Jacobi, Laguerre, Bessel and Hermite polynomials (see Appendix A, Table 1). These families are members of the class of the very classical orthogonal polynomials [8, pp. 150–152] (VCOP, in short).

Any sequence P_k = P_k(x) (k ≥ 0) from the class VCOP, has the orthogonality property, ∫_B A ϱP_kP_l dx = h_kδ_{kl}, h_k ≠ 0, (2.1)

where ϱ is the weight function, which satisfies the Pearson equation D(σϱ) = τϱ with D :=^d_{dx} , where σ and τ are polynomials, deg σ ≤ 2, deg τ = 1,

Besides three-term recurrence relation,

xP_k (x) = XP_k (x), where X := ξ_0(k) E^{−1} + ξ_1(k)I + ξ_2(k) E, any family {P_k} belonging to this class, has the following properties [8, pp. 150–152]; [15], [33, Ch. II]. First, for k ≥ 0 the second-order differential equation

L_k P_k ≡ {σD^2 + τD + λ_kI}P_k = 0 holds with λ_k := k { (1 − k)σ^{′′}/2 − τ^′ } . Second, we have the structure relation

σDP_k = λ_kDP_k , where D := δ_0(k) E^{−1} + δ_1(k)I + δ_2(k) E. Third, for any root ζ of σ, we have the differential-recurrence identity, σ(x) x − ζ Q ( λ^{−1}_k DP_k (x) ) = Pζ Pk (x), where P_ζ := I + π(ζ; k) E, Q := ϑ(k) I + ω(ζ; k) E,

where ϑ(k) := ξ_0(k)/δ_2(k).

Here I is the identity operator, Iz_k = z_k, and E^m is the mth shift operator, E^mz_k = z_{k+m} (m ∈ Z). For the sake of simplicity, we write E in place of E^1. (We adopt the convention that all the script letter operators act on the variable k).

The Fourier coefficients are defined by

a_k[f] := 1 h_k ∫_B A ϱp_kf.

2.2. Basic operators. Let {P_k } be Jacobi polynomials P^{(α,β)}_k , or Bessel polynomials Y^α_n (x) (see Table 1, in the Appendix A). Given a zero ζ of the polynomial σ associated to {P_k}, we define the following sequences of operators:

P^{(m)}_ζ := I + πm(ζ; k) E, Q^{(m)}_ζ := ϑ(k) I + ωm(ζ; k) E, R^{(m)}_ζ := δ0(k) E^{−1} + ρm(ζ; k) I, (m = 0, 1, . . .)

where

ω_m(ζ; k) := ξ_2(k + 1)π_m(ζ; k) δ_2(k + 1) , and π_m(ζ; k) := ζ

(2k + γ)_2(2k + γ + 1)_2 2(k + γ)(k + ν_{−ζ} )(2k + γ + m + 1)_2 , ρ_m(ζ; k) := −2ζ k + ν_{−ζ} + m (2k + γ + m)_2 (ζ ∈ {−1, +1})

with γ := α + β + 1, and

ν_ζ := ζ · τ(ζ) σ^{′′} = β + 1 (ζ = −1), α + 1 (ζ = +1)

(2.2)

in the Jacobi case, and

π_m(ζ; k) := −

(2k + α + 1)_2(2k + α + 2)_2 2(k + α + 1)(2k + α + m + 2)_2 , ρ_m(ζ; k) := 2 (2k + α + m + 1)_2

(ζ = 0),

in the Bessel case (notice that ζ = 0 is the double root of σ in this case).

It can be checked that

P^{(0)}_ζ = Pζ , Q^{(0)}_ζ = Qζ , R^{(0)}_ζ P^{(0)}_ζ = D, P^{(m)}_ζ Q^{(m−1)}_ζ = Q^{(m)}_ζ P^{(m−1)}_ζ (m ≥ 1), R^{(m)}_ζ P^{(m)}_ζ = P^{(m−1)}_ζ R^{(m−1)}_ζ (m ≥ 1).

Further, for i, j ≥ 0, let us define

S^{(i,j)}_ζ := { I (i < j), P^{(i)}_ζ P^{(i−1)}_ζ · · · P^{(j)}_ζ (i ≥ j), Z^{(i)}_ζ := S^{(i−1,0)}_ζ (i ≥ 0), U^{(i)}_ζ := { I (i = 0), Q^{(i−1)}_ζ · · · Q^{(1)}_ζ Q^{(0)}_ζ (i ≥ 1), M^{(i,j)}_ζ := { I (i > j), R^{(i)}_ζ R^{(i+1)}_ζ · · · R^{(j)}_ζ (0 ≤ i ≤ j), N^{(i)}_ζ := M^{(0,i−1)}_ζ .

4 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

2.3. Algorithms. Given any sequence {P_k} (k ≥ 0) of the classical orthogonal polynomials, let us introduce the differential operators U, L and Z_ζ by the following formulae: U := σD + τI, L := UD, Z_ζ := (x − ζ)D, where ζ is any root of σ.

If P_k = P^{(α,β)}_k are Jacobi polynomials^{(∗)}, let K_ζ (ζ = ±1) be the following special-form differential operator: K_ζ := (Z_ζ + ν_ζ I) D, where ν_ζ is given by (2.2).

Let assume that the function f(x) satisfies the following differential equation: P_nf(x) ≡ n∑ i=0 w_n(x)D^if(x) = g(x), (2.3)

where w_{ni} are polynomials in x, and the function g has known expansion in series of P_k .

The first step of the algorithm is to convert the LHS of the equation (2.3) to the form

P_nf = n∑ i=0 Q_i (z_if),

(2.4)

where z_i are polynomials, and the differential operators Q_i have the form Q_i := D^{pi} K^{qi}_{ζi} Z^{ri}_{ζi} L^{si} U^{ti} ,

(2.5) where ζ_i is a root of the associated polynomial σ, p_i, s_i ∈ Z^+, t_i ∈ {0, 1} and q_i ∈ {Z^+ (Jacobi case),

{0} (Bessel, Laguerre and Hermite cases),

(2.6)

r_i ∈ {0, 1} (Jacobi case),

Z^+ (Bessel case),

{0} (Laguerre and Hermite case).

(2.7)

To this end, define the differential operators P_i (i = 0, 1, . . . , n−1) and Q_j (j = 1, 2, . . . , n), and the polynomials z_0, z_1, . . . , z_n in the following recursive way.

For i = n, n − 1, . . . , 1, given the operator P_i =^i ∑ j=0 w_{ij} D^j , • represent the leading coefficient w_{ii} in the form w_{ii}(x) = [σ(x)]^{αi}(x − ζ_i)^{βi} u_i(x) where polynomial u_i has no roots in common with σ, and α_i { ∈ Z^+ (Jacobi, Bessel and Laguerre),

= i (Hermite), β_i { ∈ Z^+ (Jacobi and Bessel),

= 0 (Laguerre and Hermite), ζ_i being a root of σ (if β_i = 0, value of ζ_i is inessential), then find out which of the following cases holds:

(∗) Note that it is the only case where the associate polynomial σ(x) has two different real zeros, namely -1 and 1 (cf. Tables 4 and 5).

– case A: α_i ≥ m, – case B: α_i + β_i ≥ m > α, – case C: m > α_i + β_i, where m := ⌊(i + 1)/2⌋;

• define

z_i(x) := u_i(x) ·

[σ(x)]^{αi−m}(x − ζ)^{βi} (case A), (x − ζ)^{αi+βi−m} (case B), 1 (case C);

• define

r_i := i mod 2 (case B), 0 (case A or C), t_i := i mod 2 (case A), 0 (case B or C), s_i := min{α_i, ⌊i/2⌋} q_i := min{β_i, ⌊i/2⌋ − t_i}, p_i := i − 2q_i − 2s_i − r_i − t_i; • define the operator Q_i by Q_i := D^{pi} K^{qi}_{ζi} Z^{ri}_{ζi} L^{si} U^{ti} ;

• define the operator P_{i−1} of degree i − 1 as P_{i−1}f(x) := P_if(x) − Q_i (z_if)(x).

For convenience, let us denote Q_0 := I, z_0 := w_{0,0},

(2.8) where w_{0,0} is the only coefficient of the operator P_0.

Algorithm 2.1 ([26]). Let {P_k } be any sequence of the very classical orthogonal polynomials, and let the function f obey the differential equation P_nf(x) ≡ n∑ i=0 Q_i (z_if)(x) = g(x),

(2.9)

where z_i are polynomials, and the ith-order differential operator Q_i (i = 0, 1, . . . , n) is of the form Q_i := D^{pi} K^{qi}_ζ Z^{ri}_ζ L^{si} U^{ti} ,

(2.10) where ζ is a fixed root of the associated polynomial σ, p_i, s_i ∈ Z^+, t_i ∈ {0, 1} and q_i ∈ {Z^+ (Jacobi case),

{0} (Bessel, Laguerre and Hermite cases),

r_i ∈ {0, 1} (Jacobi case),

Z^+ (Bessel case),

{0} (Laguerre and Hermite case). Define e := max_{0≤i≤n} p_i, d := max_{0≤i≤n}(p_i + 2q_i + r_i) − e,

T := Z^{(d)}_ζ D^e, and Ai := { μ_{qi} (ζ; k) E^{qi} U^{(ri)}_ζ (−λ_k)^{si+ti} D^{ti} (Jacobi and Bessel), (−λ_k)^{si+ti} D^{ti} (Laguerre and Hermite),

6 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

Ci := { M^{(d,d+e−pi−1)} ζ S^{(d+e−pi−1,2qi+ri)} ζ (Jacobi and Bessel),

D^{e−pi} (Laguerre and Hermite),

where i = 0, 1, . . . , n.

Then the Fourier coefficients a_k[f] satisfy the recurrence relation L(hk ak[f]) = B(k), (2.11) where

L :=^{n∑}_{i=0} CiAizi(X),

(2.12)

B(k) := T(h_ka_k [g]).

(2.13)

In the Jacobi case, it is possible that not all ζ_i’s in (2.5) are equal, so that Algorithm 2.1 is not applicable. The next algorithm is a corrected version of a result in [19]. Algorithm 2.2 ([26]). Let {P_k} be the sequence of the Jacobi polynomials, and let f satisfies the equation P_nf(x) ≡ n∑ i=0 Q_i (z_if)(x) = g(x),

where z_i are polynomials, and the differential operators Q_i have the form Q_i := D^{pi} K^{qi}_{ζi} Z^{ri}_{ζi} L^{si} U^{ti} , where ζ_i ∈ {−1, +1}, p_i, q_i, s_i ∈ Z^+, r_i, t_i ∈ {0, 1}, p_i +2q_i +r_i +2s_i +t_i = i, and the expansion of the function g in P_k is known.

Let us introduce the notation Ω := {1, 2, . . . , n}, Ω_η := {i ∈ Ω : ζ_i = η} (η ∈ {−1, +1}), v_i := 2q_i + r_i (i ∈ Ω), e_η := max_{i∈Ωη} p_i, d_η := max_{i∈Ωη} (p_i + v_i) − e_η, (η ∈ {−1, +1}),

ω := { −1 if d_{−1} + e_{−1} ≥ d_1 + e_1, +1 if d_{−1} + e_{−1} < d_1 + e_1, e := max(e_ω , d_{−ω} + e_{−ω}), d := d_ω + e_ω − e.

Further, let define difference operators T, A_i, B_i, C_i, J_η by

T := Z^{(d)}_ω D^e, (∗) Ai := μq_i (ω; k) E^{qi} U^{(ri)}_ω (−λk )^{si+ti} D^{ti} (i = 0, 1, . . . , n), Bi := M^{(dη,eη +dη−pi−1)} η S^{(eη +dη−pi−1,vi)} η (i ∈ Ω_η; η = −1, +1), J_ω := M^{(d,dω−1)}_ω , J_{−ω} := Z^{(d)}_ω N^{(e−e−ω )}_{−ω} S^{(e−e−ω −1,d−ω)}_{−ω} , C0 := T, Ci := { J_ωBi (i ∈ Ωω), J−ωBi (i ∈ Ω−ω). The Fourier coefficients a_k[f] satisfy the recurrence relation L(hkak [f]) = B(k), where

L :=^{n∑}_{i=0} CiAizi(X),

B(k) := T(h_ka_k [g]).

3. Discrete classical and q-classical orthogonal polynomials 3.1. Properties of the discrete and q-classical classical orthogonal polynomials. Let {P_k} be any member of the family of classical orthogonal polynomials of a discrete variable (DCOP, in short) [34, Chapter II], or q-classical polynomials in the Hahn sense (qCOP, in short) [13, 16, 31, 32], See Appendix A, Tables 2 and 3.

We introduce the notation E^±f(x) := f(x ± 1), D^+ := Δ, D^− := ∇ (DCOP), E^±f(x) := f(xq^{±1}), D^+ := D_q, D^− := D_{1/q} (qCOP).

Here the forward and backward difference operators Δ and ∇ are defined by Δf(x) := f(x + 1) − f(x), ∇f(x) := f(x) − f(x − 1), and the q-derivative operator D_q is given by

D_qf(x) := f(qx) − f(x) (q − 1)x (x ≠ 0), with D_qf(0) := f^′(0) (provided f^′(0) exists).

Any system {P_k} of the classical orthogonal polynomials of a discrete variable, or q-classical polynomials, has the orthogonality property, BS_A {ϱPkPl} dx = hk δkl, hk ≠ 0, (3.1)

where ϱ is the weight function, which satisfies the Pearson equation D^+ (σϱ) = τϱ,

where σ and τ are polynomials, deg σ ≤ 2, deg τ = 1. Here the symbol^BS_A dx stands for summation ∑_B x=A (case DCOP), or q-integral ∫_B A d_qx (case qCOP).

Also, we have the following three-term recurrence relation, structure relations, difference equation and difference-recurrence identities [15], [29], [34]:

xP_k(x) = XP_k(x), where X := ξ_0(k) E^{−1} + ξ_1(k)I + ξ_2(k) E, L_kP_k ≡ {σD^+ D^− + τD^+ + λ_kI}P_k = 0, λ_k := κ(k) [ κ(1 − k)σ^{′′}/2 − τ^′ ] ; (3.2)

σ^±D^±P_k = λ_kD^±P_k , where D := δ^±_0 (k) E^{−1} + δ^±_1 (k)I + δ^±_2 (k) E,

(3.3) σ^±(x) x − ζ^± Q ± ζ± ( λ^{−1}_k D^±P_k(x) ) = P^±_ζ± Pk(x), where ζ^± is any root of σ^±, and P^±_ζ± := I + π^±(ζ^±; k) E, Q^±_ζ± := ϑ^±(k) I + ω^±(ζ^±; k) E.

Here ϑ^±(k) := ξ_0(k)/δ^±_0 (k),

κ(m) := m (DCOP), q^m − 1 q − 1 (qCOP), σ^− := σ, σ^+ := { σ + τ (DCOP), σ + x(q − 1)τ (qCOP).

(3.4)

The Fourier coefficients are now defined by

a_k[f] := 1 h_k BS_A {ϱpkf } dx. (3.5)

8 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

3.2. Basic operators. Given roots ζ^+ and ζ^− of the polynomials σ^+ and σ^−, respectively, let us define for i = 0, 1, . . . the operators P^{±(i)}_{ζ±} and Q^{±(i)}_{ζ±} of the form

P^{±(i)}_ζ± := I + π^±_i (ζ^±; k) E, Q^{±(i)}_ζ± := ϑ^±(k) I + ω^±_i (ζ^±; k) E,

(3.6)

in the following recursive way:

π^±_0 (ζ; k) := π^±(ζ; k), ω^±_0 (ζ; k) := ω^±(ζ; k), π^±_i (ζ; k) := η^±(k + i) ω^±_i (ζ^±; k), ω^±_i (ζ^±; k) :=

1 − ϑ^±(k)η^±(k + i − 1) 1 − ϑ^±(k + 1)η^±(k + i) ω^±_{i−1}(ζ^±; k) (i ≥ 1), where η^±(k) := δ^±_2 (k + 1)/ξ_2(k + 1).

It can be checked that

P^{±(0)}_ζ± = P^±_ζ± , Q^{±(0)}_ζ± = Q^±_ζ± , P^{±(i)}_ζ± Q^{±(i−1)}_ζ± = Q^{±(i)}_ζ± P^{±(i−1)}_ζ± (i = 1, 2, . . .).

Further, let S^{±(i,j)}_ζ± := { I (i < j), P^{±(i)}_ζ± P^{±(i−1)}_ζ± · · · P^{±(j)}_ζ± (i ≥ j ≥ 0) Z^{±(i)}_ζ± := S^{±(i−1,0)}_ζ± (i ≥ 0), U^{±(0)}_ζ± := I, U^{±(i)}_ζ± := Q^{±(i−1)}_ζ± . . . Q^{±(1)}_ζ± Q^{±(0)}_ζ± (i ≥ 1).

Having defined the sequence of operators P^{±(i)}_{ζ±} (i = 0, 1, . . .) corresponding to a given root ζ^±, let us define for i = 0, 1, . . . the operators R^{±(i)}_{ζ±} of the form

R^{±(i)}_ζ± := δ^±_0 (k) E^{−1} + ρ^±_i (ζ; k) I,

(3.7) where δ^±_0 is that of (3.3), in the following recursive way:

ρ^±_0 (ζ^±; k) := δ^±_2 (k) π^±(ζ^±; k) , ρ^±_i (ζ^±; k) := π^±_{i−1}(ζ^±; k) π^±_i (ζ^±; k) ρ^±_{i−1}(ζ^±; k + 1) (i ≥ 1).

It can be checked that

R^{±(0)}_ζ± P^{±(0)}_ζ± = D^±, R^{±(i)}_ζ± P^{±(i)}_ζ± = P^{±(i−1)}_ζ± R^{±(i−1)}_ζ± (i = 1, 2, . . .).

Further, for i, j ≥ 0, let us define M^{±(i,j)}_ζ± := { I (i > j), R^{±(i)}_ζ± R^{±(i+1)}_ζ± · · · R^{±(j)}_ζ± (0 ≤ i ≤ j), N^{±(i)}_ζ± := M^{±(0,i−1)}_ζ± .

Finally, let us define the operators F^{±(i)}_{ω+,ω−} (i = 0, 1, . . . ; ω^± ∈ {ζ^±, ζ^±_∗ }) of the form

F^{±(i)}_ω+_{,ω}− := φ^±(ω^+, ω^−; k) E^{−1} + φ^±_i (ω^+, ω^−; k) I,

where

φ^±(ω^+, ω^−; k) := ε^± ν^±_{ω+,ω−} δ^∓_0 (k) γ^±(ω^+, ω^−; k − 1) , and γ^±(ω^+, ω^−; k) := ϑ^±(k) + ν^±_{ω+,ω−} ,

(3.8) ν^±_{ω+, ω−} := 2 (σ^±)^{′′} τ(ω^∓_∗ ) ω^∓_∗ − ω^±_∗ , (3.9)

in the following recursive way:

φ^±_0 (ω^+, ω^−; k) := ε^± ν^±_{ω+,ω−} ρ^∓_0 (ω^∓; k) γ^±(ω^+, ω^−; k) , φ^±_i (ω^+, ω^−; k) := π^±_{i−1}(ω^±; k) π^±_i (ω^±; k) φ^±_{i−1}(ω^+, ω^−; k + 1) (i ≥ 1).

Here

ε^+ := 1, ε^− := { 1 (DCOP), q (qCOP).

Obviously, the sequence F^{±(i)}_{ω+,ω−} is well-defined only in the case when ω^+_∗ ≠ ω^−_∗ (cf. (3.9), (3.8)).

It can be verified that

F^{±(0)}_ω+_{,ω}− := ε^± ν^±_ω+_{,ω}− R^{∓(0)}_ω∓^{( 1}_{γ±(ω+, ω−; k)} I^) , F^{±(i)}_ω+_{,ω}− P^{±(i)}_ω± = P^{±(i−1)}_ω± F^{±(i−1)}_ω+_{,ω}− (i = 1, 2, . . .),

where we use notation of (3.6) and (3.7). Further, let

W^{±(i,j)}_ω+_{,ω}− := { I (i > j), F^{±(i)}_ω+_{,ω}− F^{±(i+1)}_ω+_{,ω}− · · · F^{±(j)}_ω+_{,ω}− , (0 ≤ i ≤ j), Y^{±(i)}_ω+_{,ω}− := W^{±(0,i−1)}_ω+_{,ω}− (i ≥ 0).

3.3. Algorithm. Let {P_k} be any sequence of the classical orthogonal polynomials of a discrete variable, or q-classical polynomials. Let us define the difference operators U^± and Z^±_{ζ±} (acting on x) by

U^± := 1 ε^± σ^±(x)D^± + τ(x)I,

Z^±_{ζ±} := 1 ε^± (x − ζ^±)D^± ,

where ζ^± is any root of the polynomial σ^±.

Algorithm 3.1 ([28, 48]). Let {P_k } be any sequence of the classical orthogonal polynomials of a discrete variable, or q-classical polynomials in the Hahn sense. Let assume that the function f(x) satisfies the following equation: Pf(x) ≡ P^+f(x) + P^−f(x) = g(x),

(3.10)

10 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

where

P^±f(x) := n±∑ i=0 w^±_{n±,i}(x) ( D^± )_i f(x), (3.11)

and w^±_{n±,i} are polynomials.

Step 1: With n := n^+, define the differential operators P^+_i (i = 0, 1, . . . , n − 1) and Q^+_j (j = 1, 2, . . . , n), and the polynomials z^+_0 , z^+_1 , . . . , z^+_n in the following recursive way.

For i = n, n − 1, . . . , 1, given the operator P^+_i =^i ∑ j=0 w^+_{ij} (D^+ )^j , • represent the leading coefficient w^+_{ii} in the form

w^+_{ii} (x) = (_{αi} ∏ h=1 ( E^+ )_{i−h} σ^+(x) ) · (_{βi} ∏ m=αi+1 ( E^+ )_{i−m} (x − ζ^+_i ) ) · u_i(x) where u_i is a polynomial, ζ^+_i is a root of σ^+, and the integer exponents α_i, β_i have maximal possible values such that 0 ≤ α_i ≤ i, and α_i ≤ β_i ≤ B_i, where B_i = i if deg σ^+ = 2, and B_i = α_i if deg σ^+ < 2 (if β_i = α_i, the value of ζ_i is inessential);

• define r^+_i := i − β_i, s^+_i := β_i − α_i, t^+_i := α_i, z^+_i (x) := ( E^+ )_{−i} u_i(x);

• define the operator Q^+_i by

Q^+_i := ( D^+

)_r+_i (

Z^+_{ζ+i}

)_s+_i ( U^+

)_t+_i ; • define the operator P^+_{i−1} of degree i − 1 as P^+_{i−1}f(x) := P^+_i f(x) − Q^+_i (z^+_i f)(x).

Denote Q^+_0 := I, z^+_0 := w_{0,0}.

Step 2: Repeat Step 1 with the superscript ’+’ changed for ’−’. Step 3: With n := n^+, ζ := ζ^+, ζ_∗ := ζ^+_∗ , ζ_i := ζ^+_i , r_i := r^+_i , s_i := s^+_i , t_i := t^+_i , z_i := z^+_i , define Ω := {1, 2, . . . , n}, Ω_η := {i ∈ Ω : ζ_i = η} (η ∈ {ζ, ζ_∗}), e_η := max_{i∈Ωη} r_i, d_η := max_{i∈Ωη} (r_i + s_i) − e_η (η ∈ {ζ, ζ_∗}),

ω^+ := ζ, ω^+_∗ := ζ_∗ if d_ζ + e_ζ ≥ d_{ζ∗} + e_{ζ∗} , ω^+ := ζ_∗, ω^+_∗ := ζ if d_ζ + e_ζ < d_{ζ∗} + e_{ζ∗} , e^+ := max(e_ω+ , d_ω+_∗ + e_ω+_∗ ), d^+ := d_ω+ + e_ω+ − e^+.

Further, define

T^+ := Z^{+(d+)}_ω+^(D^{+)e+} , Ai :=^(ε^{+)ri} U^{+(si)}_ω+^(−^{λk}_{ε−} D^{−)ti} (i = 0, 1, . . . , n), Bi := M^{+(dη,eη+dη −ri−1)} η S^{+(eη +dη−ri−1,si)} η (i ∈ Ω_η; η = ζ, ζ_∗),

J_ω+ := M^{+(d,d+ω −1)}

ω+ , J_ω+_∗ := Z^{+(d)}_{ω+} N^{+(e−eω+∗ )}_{ω+∗} S^{+(e−eω+∗ −1,dω+∗ )} ω+∗ , C0 := T^+, Ci := { J_ω+ Bi (i ∈ Ω^+_ω ), J_ω+_∗ Bi (i ∈ Ω_ω+_∗ ), L^+ :=^{n∑}_{i=0} CiAizi(X).

Step 4: Repeat Step 3 with the superscript ’+’ changed for ’−’, and vice versa. Step 5: If the conditions deg σ = deg σ^+ = 2, and (ζ^+ − ζ^−_∗ )(ζ^− − ζ^+_∗ ) ≠ 0 are satisfied, go to Step 6. Otherwise, Step 5a: if d^+ ≥ d^−, replace the operators T^− and L^− by

T^− :=^(D^{−)d−+e−} , L^− := N^{−(d−)}_ζ− L^−,

and set e^− := d^− + e^−, d^− := 0, while Step 5b: if d^+ < d^−, replace the operators T^+ and L^+ by

T^+ :=^(D^{+)d++e+} , L^+ := N^{+(d+)}_ζ+ L^+,

and set e^+ := d^+ + e^+, d^+ := 0.

Step 6: Define

T := ZY^{+(δ)}_{ω+,ω}−_∗ Z^{+(δ)}_ω+^(D^{−)e− (}D^{+)e+} (δ := min(d^+, d^−)), (∗∗) L = C^+L^+ + C^−L^−,

where

Z := { Z^{+(d+−d−)} ω+ (d^+ ≥ d^−), Z^{−(d−−d+)} ω− (d^+ < d^−), C^+ := W^{+(d+−d−,d+−1)} ω+,ω−∗ ( Δ^{+(d+)} ω+ )_e− (d^+ ≥ d^−),

ZY^{+(d+)}_{ω+,ω}−_∗^(Δ^{+(d+)}_ω+^{)e−} (d^+ < d^−),

C^− := (ε^−)^{e+e−} ZY^{−(d−)}_{ω+∗ ,ω−} ( Δ^{−(d−)} ω− )_e+ (d^+ ≥ d^−), (ε^−)^{e+e−} W^{−(d−−d+,d−−1)} ω+∗ ,ω− ( Δ^{−(d−)} ω− )_e+ (d^+ < d^−).

Here Δ^{±(i)}_{ω±} := R^{±(i)}_{ω±} P^{±(i)}_{ω±} ± I (i ≥ 0), in the case DCOP, and

Δ^{±(i)}_{ω±} :=

D^∓ (i = 0), [ P^{±(i−1)}_ω± ± (q − 1)Q^{±(i−1)}_ω±^] R^{±(i−1)}_ω± ± ω^±(q − 1) I (i ≥ 1),

12 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

in the case qCOP.

Then the coefficients a_k[f] satisfy the recurrence relation L(hkak[f]) = φ(k), where

L := C^+L^+ + C^−L^−

φ(k) := T(h_k a_k[g]) .

4. Examples 4.1. Very classical orthogonal polynomials. Let the fuction f(x) = (1 + x)e^{−x−1} be expanded in the Jacobi polynomials P^{(α,β)}_k , f = ∞∑ k=0 a_k[f]P^{(α,β)}_k . This function satisfies the first-order differential equation (x + 1)f^′(x) + xf(x) = 0.

A Maple session initialized by

>read(‘jacobi.mpl‘):
>Jacobi([x,x+1]);

brings the following results:

A[−1] = −(2 k + 5 + α + β) (2 k + α + β + 2) (2 k + α + β + 1) (2 k + α + β) (2 k + 4 + α + β)^2 (2 k + α + β + 3)^2 A[0] = −(kα^2 − α^2 + 2kαβ + 3kα + 4k^2α − α + 4k^2β + 4k^2 + kβ^2 − 2k + β^2 + 4k^3 − β + kβ) (2k + α + β + 3)(2k + 5 + α + β)(2k + α + β + 2)(2k + α + β + 1)(2k + 4 + α + β)^2 A[1] = −2(α^3 + 10α^2 + 3βα^2 + 5kα^2 + 10kαβ + 16αβ + 27kα + 3αβ^2 + 21α +8k^2α + 6β^2 + β^3 + 5kβ^2 + 25kβ + 12 + 20k^2 + 4k^3 + 8k^2β + 17β + 30k) (2k + 4 + α + β)(k + 1)(2k + α + β + 3)(k + β + 1)(2k + 5 + α + β)(2k + α + β) A[2] = 8(k + 2 + α)(k + α + β + 2)(k + 1)(k + 2)(k + β + 1)(k + 2 + β)(2k + α + β + 1) (2k + α + β) w = (2 k + 5 + α + β) (2 k + α + β + 2) (2 k + α + β + 1) (2 k + α + β) (2 k + 4 + α + β)^2 (2 k + α + β + 3)^2

s = 0, ω = −1, d = 1, e = 0 T = E^s Z_ω^{(d)} D^e h(k) [−1, 2, A, (2 k + 5 + α + β) (2 k + α + β + 2) (2 k + α + β + 1) (2 k + α + β) (2 k + 4 + α + β)^2 (2 k + α + β + 3)^2] Thus, we have obtained the third-order recurrence relation 2∑ i=−1 A[i]a_{k+i}[f] = 0

with the above given coefficients A[i].

4.2. Classical orthogonal polynomials of a discrete variable. Let f(x) = (x − n + 1)_n (Pochhammer’s polynomial) be represented in terms of the Hahn polynomials, f(x) = n∑ k=0 a_k[f]Q_k(x; α, β, N) (n < N). This function satisfies the first-order difference equation xD^− f(x) − nf(x) = 0 with D^− f(x) ≡ ∇f(x) = f(x) − f(x − 1).

A Maple session initialized by

>read(‘hahn.mpl‘):
>Hahn([-n,x],[]);

yields A[0] = (−n + k) (2 k + α + β + 2) (2 k + α + β + 3) A[1] = −(n + 2 + k + α + β) (k + 1) (k − N + 1) (k + β + 1) w = (2 k + α + β + 2) (2 k + α + β + 3) s = 0, ε = 0, ζ = N − 1, η = N + α, d = 1, t = 0, u = 0, v = 0 T = E^s Z_ε^{−(d)} Y_{ζ, η}^{+(t)} Z_ζ^{+(t)} (D^−)^u (D^+)^v h(k) [0, 1, A, (2 k + α + β + 2) (2 k + α + β + 3)] Thus, we have obtained the first-order recurrence relation A[0]a_k[f] + A[1]a_{k+1}[f] = 0 with the above given coefficients A[0], A[1].

4.3. q-Classical orthogonal polynomials. Let the polynomial f(x) = x^n be represented in terms of the little q-Jacobi polynomials, f(x) = n∑ k=0 a_k[f]p_k(x; a, b|q). This function satisfies the first-order q-difference equation xD^+ (f)(x) − q^n − 1 q − 1 f(x) = 0

with D^+ ≡ D_q . A Maple session started by the commands

>read(‘l_jacobi.mpl‘):
>l_qJacobi([],[-qnum(n),x]);

yields

A[0] = − q^n − q^k q − 1 A[1] =

(a q^k q − 1) (q^k q − 1) q^k (a b q^k q^2 q^n − 1) (a b (q^k)^2 q^3 − 1) (q − 1) (a b (q^k)^2 q^2 − 1) s = 0, ζ = 0, η = 0, d = 1, t = 0, u = 0, v = 0 T = E^s Z_ζ^{+(d)} Y_{ζ, η}^{+(t)} Z_ζ^{+(t)} (D^−)^u (D^+)^v h(k)

14 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY [0, 1, A] Thus, we have obtained the first-order recurrence relation A[0]a_k[f] + A[1]a_{k+1}[f] = 0 with the above given coefficients A[0], A[1].

Appendix A. Data of the classical orthogonal polynomials

Table 1. Hypergeometric series representations of the monic very classical orthogonal polynomials

Family Hypergeometric series

Jacobi P^{(α,β)}_k (x) = (−1)^k ( k + β k ) 2F_1 ( −k, k + α + β + 1 β + 1 | | | 1 + x 2 )

Laguerre L^α_k (x) = (α + 1)_k(−1)^k_1F_1 ( −k α + 1 | | | x )

Hermite H_k(x) = x^k_2F_0 ( −k/2, −k/2 + 1/2 — | | | − 1 x^2 )

Bessel Y^α_k (x) =

2^k (k + α + 1)_k^2

F_0 ( −k, k + α + 1 — | | | − x 2 )

Table 2. Hypergeometric series representations of the monic classical orthogonal polynomials of a discrete variable

Family Hypergeometric series

Hahn Q_k (x; α, β, N) =

(β + 1)_k(1 − N)_k (k + α + β + 1)_k^3

F_2 ( −k, −x, k + α + β + 1 β + 1, 1 − N | | | 1 )

Meixner M^{(β,c)}_k (x) = (β)_k ( c c − 1 )_k 2F_1 ( −k, −x β | | | 1 − 1 c )

Krawtchouk K^{(p)}_k (x, N) = (−N)_kp^k ( N k ) 2F_1 ( −k, −x −N | | | − 1 p )

Charlier C^{(a)}_k (x) = (−a)^k_2F_0 ( −k, −x — | | | | − 1 a )

Table 3. Basic hypergeometric series representations of the monic q-classical orthogonal polynomials [11, 15, 32]

Family Basic hypergeometric series

big q-Jacobi P_k(x; a, b, c; q) = (aq; q)_k(cq; q)_k (abq^{k+1}; q)_k^3 ϕ_2 ( q^{−k}, abq^{k+1}, x aq, cq | | q; q )

big q-Laguerre P_k(x; a, b; q) = (aq; q)_k(bq; q)_{k 3}ϕ_2 ( q^{−k}, 0, x aq, bq | | q; q )

q-Krawtchouk K_k(x; p, N ; q) =

(q^{−N} ; q)_k (−pq^{−k}; q)_k^3 ϕ_2 ( q^{−k}, −pq^k, x q^{−N} , 0 | | q; q )

little q-Jacobi p_k(x; a, b|q) = (−1)^kq( k 2) (aq; q)^k (abq^{k+1}; q)_k^2 ϕ_1 ( q^{−k}, abq^{k+1} aq | | q; qx )

q-Meixner M_k(x; b, c; q) = (−c)^kq^{−k2}(bq; q)_{k 2}ϕ_1 ( q^{−k}, x bq | | q; − q^{k+1} c )

alternative q-Charlier K_k(x; a; q) = (−1)^kq( k 2)(aq; q)−1_{k 2}ϕ_1 ( q^{−k}, −aq^k 0 | | q; qx )

little q-Laguerre/Wall L^{(a)}_k (x; q) = (−1)^kq( k 2)(aq; q)_{k 2}ϕ_1 ( q^{−k}, 0 aq | | q; qx )

Al-Salam&Carlitz I U^{(a)}_k (x; q) = (−a)^kq( k 2)_2ϕ_1 ( q^{−k}, 1/x 0 | | q; qx/a )

discrete q-Hermite h_k(x; q) = q( k 2)_2ϕ_1 ( q^{−k}, 1/x 0 | | q; −qx )

Al-Salam&Carlitz II V^{(a)}_k (x; q) = (−a)^kq^−( k 2)_2ϕ_0 ( q^{−k}, x − | | q; q^k a )

Stieltjes-Wigert S_k(x; q) = (−1)^kq^{−k2}_1ϕ_1 ( q^{−k} 0 | | q; −q^{k+1}x )

16 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

Table 4. Data for the monic Jacobi and Bessel polynomials

Jacobi Bessel

σ(x) x^2 − 1 x^2

τ (x) (γ + 1)x + δ (α + 2)x + 2 λ_k −k(k + γ) −k(k + α + 1) h_k 2^{2k+γ} k!Γ(k + α + 1)Γ(k + β + 1) Γ(2k + γ + 1)(k + γ)_k

(−1)^{k+1}2^{2k}k!

(α + 1)_{2k+1}(k + α + 1)_k ξ_0(k)

4k(k + α)(k + β)(k + γ − 1) (2k + γ − 2)_3(2k + γ − 1) −4k(k + α) (2k + α)(2k + α − 1)_3 ξ_1(k) − (γ − 1)δ (2k + γ − 1)(2k + γ + 1) − 2α (2k + α)(2k + α + 2) ξ_2(k) 1 1

δ_0(k)

4(k + α)(k + β)(k + γ − 1) (2k + γ − 2)_3(2k + γ − 1) − 4(k + α) (2k + α)(2k + α − 1)_3 δ_1(k) 2δ (2k + γ − 1)(2k + γ + 1) 4 (2k + α)(2k + α + 2) δ_2(k) − 1 k + γ − 1 k + α + 1 Note: γ := α + β + 1, δ := α − β.

Table 5. Data for the monic Laguerre and Hermite polynomials

Laguerre Hermite σ(x) x 1 τ (x) 1 + α − x −2x λ_k k 2k h_k k!Γ(k + α + 1) √ π2^{−k}k! ξ_0(k) k(k + α) k 2 ξ_1(k) 2k + α + 1 0 ξ_2(k) 1 1 δ_0(k) k + α 1 2 δ_1(k) 1 0 δ_2(k) 0 0

Table 6. Data for the monic Hahn polynomials

σ(x) x(N + α − x) τ (x) (β + 1)(N − 1) − (γ + 1)x σ^+ (x) (x + β + 1)(N − 1 − x) λ_k k(k + γ) h_k k!Γ(k + α + 1)Γ(k + β + 1)(2k + γ + 1)_{N−k−1} (k + γ)_k (N − k − 1)!

ξ_0(k) k(N − k)(k + α)(k + β)(k + γ − 1)(k + γ + N − 1) (2k + γ − 2)_2(2k + γ − 1)_2 ξ_1(k) α − β + 2N − 2 4 + (β^2 − α^2)(γ + 2N − 1) 4(2k + γ − 1)(2k + γ + 1) ξ_2(k) 1 δ^+_0 (k) = δ^−_0 (k)

(k + α)(k + β)(k + γ − 1)(N − k)(k + γ + N − 1) (2k + γ − 2)_2(2k + γ − 1)_2 δ^+_1 (k) = δ^−_1 (k) − 1 −

[2k(k + γ) + (γ − α)(γ − 1) − N (α − β)] (2k + γ − 1)(2k + γ + 1) δ^+_2 (k) = δ^−_2 (k) − 1 k + γ ω^±(ζ^±; k) δ^±_1 (k)ξ_0(k)/δ^±_0 (k) + ζ^± − ξ_1(k) ξ_0(k + 1)δ^±_2 (k + 1)/ξ_2(k + 1) − δ^±_0 (k + 1) π^±(ζ^±; k) δ^±_2 (k + 1)ω^±(ζ^±; k)/ξ_2(k + 1) ρ^±(ζ^±; k) δ^±_2 (k)/π^±(ζ^±; k) Notes: γ := α + β + 1; ζ^+ (resp., ζ^−) is any root of the polynomial σ^+ (resp., σ).

18 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

Table 7. Data for the monic Charlier, Meixner and Krawtchouk polynomials

Charlier Meixner Krawtchouk σ(x) x x x τ (x) a − x βc + (c − 1)x (1 − p)^{−1}(N p − x) σ^+ (x) a c(x + β) p(1 − p)^{−1}(N − x) λ_k k (1 − c)k (1 − p)^{−1}k h_k k!a^k k!(β)_kc^k (1 − c)^{β+2k} N !k! (N − k)! p^k(1 − p)^k ξ_0(k) ak ck(k + β − 1) (1 − c)^2 p(1 − p)k(N − k + 1) ξ_1(k) k + a

(c + 1)k + βc 1 − c k + p(N − 2k) ξ_2(k) 1 1 1 δ^+_0 (k) = δ^−_0 (k) a c(k + β − 1) (1 − c)^2 p(1 − p)(N + 1 − k) δ^+_1 (k) = δ^−_1 (k) − 1 0 c 1 − c −p δ^+_2 (k) = δ^−_2 (k) 0 0 0

Table 8. Data of the q-classical orthogonal polynomials: coefficients σ and τ of the equation (3.2), and σ^+ (cf. (3.4)) [31, 32]

p_k(x; q) σ(x) τ (x) σ^+ (x) P_k(x; a, b, c; q) (big q-Jacobi) (x − aq)(x − cq) abq^2(x − 1) − x + cq − aq(cq − 1) q − 1 aq^2(x − 1)(bx − c)

P_k(x; a, b; q) (big q-Laguerre) (x − aq)(x − bq) bq − x − aq(bq − 1) q − 1 abq^2(1 − x)

K_k(x; p, N ; q) (q-Krawtchouk) x(x − q^{−N} ) pq(1 − x) − x + q^{−N} q − 1 pqx(1 − x)

p_k(x; a, b|q) (little q-Jacobi) x(x − 1) aq(bqx − 1) + 1 − x q − 1 aqx(bqx − 1)

M_k(x; b, c; q) (q-Meixner) c(x − bq) q(x − 1) + c(bq − 1) q − 1 q(x − 1)(x + bc)

K_k(x; a; q) (alternative q-Charlier) x(x − 1) 1 − (aq + 1)x q − 1 −aqx^2

L^{(a)}_k (x; q) (little q-Laguerre/Wall) x(x − 1) x − 1 + aq q − 1 −aqx

U^{(a)}_k (x; q) (Al-Salam&Carlitz I ) (x − 1)(x − a) x − a − 1 1 − q a

h_k(x; q) (discrete q-Hermite ) x^2 − 1 x 1 − q −1

V^{(a)}_k (x; q) (Al-Salam&Carlitz II ) a x − a − 1 q − 1 (x − 1)(x − a)

S_k(x; q) (Stieltjes-Wigert) x qx − 1 q − 1 qx^2

20 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

Table 9. Data of the monic q-classical orthogonal polynomials: coefficients of the difference operators D^±, X, P^±_ζ± and Q^±_ζ± given in terms of σ and τ [29, 31]

ξ_0(k) q^{k−1}[k]_qR(k − 2){s_0R(2k − 2) − q^{k−1}(s_1[k − 1]_q + t_0) ( q^{k−1}s_2 t_0 − s_1R(k − 1) ) } R(2k − 1)R^2(2k − 2)R(2k − 3)

ξ_1(k) q^k s_2q^{k−1}[k]_q − R(k − 2) − q^{−1}s_1[2]_q[k]_qR(k − 1) R(2k)R(2k − 2) ξ_2(k) 1

δ^+_0 (k) −q^{1−k}R(k − 1)ξ_0(k)/λ_k

δ^+_1 (k) − qt_0 (q + 1)λ_k − qξ_1(k){(q − 1)[−k]_qR(k − 1) + t_1} (q + 1)λ_k

δ^+_2 (k) s_2[k]_q/λ_k

δ^−_i (k) δ^+_i (k) + (q − 1)ξ_i(k) (i = 0, 1, 2) ϑ^+(k) −q^{k−1}λ_k/R(k − 1) ϑ^−(k) 1/[q − 1 − q^{1−k}R(k − 1)/λ_k] ω^±(ζ^±; k) δ^±_1 (k)ξ_0(k)/δ^±_0 (k) + ζ^± − ξ_1(k) ξ_0(k + 1)δ^±_2 (k + 1)/ξ_2(k + 1) − δ^±_0 (k + 1)

π^±(ζ^±; k) δ^±_2 (k + 1)ω^±(ζ^±; k)/ξ_2(k + 1)

ρ^±(ζ^±; k) δ^±_2 (k)/π^±(ζ^±; k)

Notes: 1. t_1 := τ^′, t_0 := τ (0), s_2 := σ^{′′}/2 + (q − 1)τ^′, s_1 := σ^′(0) + (q − 1)τ (0), s_0 := σ(0), R(k) := s_2[k]_q + t_1.

2. ζ^+ (resp., ζ^−) is any root of the polynomial σ^+ (x) := σ(x) + (q − 1)xτ (x) (resp., σ^− ≡ σ).

Appendix B. Maple programs

B.1. Expansions in the Jacobi polynomials.

# Constructing recurrence equation for Jacobi series coefficients
# of a solution of a linear differential equation.
#
###################################################################
#
# Version 1.0 for MapleV release 4
#
# Pawel Wozny (Pawel.Wozny@ii.uni.wroc.pl), 07/02/2001
#
###################################################################
#
# Let f(x) be a solution of a linear differential equation
#
# n
# -----
# \ i
# (1) ) v_i(x) D f(x) = g(x),
# /
# -----
# i = 0
# d
# where v_i(x) are polynomials, and D:= --.
# dx
# Let a_k[f] be the Fourier coefficients of the function f(x) in the
# expansion
# -----
# \
# (2) f(x) = ) a_k[f] P_k(x),
# /
# -----
# k
#
# where P_k(x) is the k-th monic Jacobi polynomial. This program finds
# recurrence relation for a_k[f] of the form
#
# r + d
# -----
# \
# (3) ) A_i(k) a_{i+r}[f] = B(k).
# /
# -----
# i = -r
#
# under assumption that a_k[g] are known explicitly. Here r is a natural
# number and d {0,1}.
#
# Main procedure
# ~~~~~~~~~~~~~~
# - Jacobi(V)
#
# Input
# ~~~~~
# - vector V := [v_0(x),v_1(x),...,v_n(x)]
#
# Output
# ~~~~~~
# - vector [-r,r+d,A,w], where A[-r..r+d] is a table of A_l(k)

22 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

# (l=-r..r+d)
# - operator T (see (*)) is given on the screen such that
#
# B(k)= w*T(a_k[g])
#
###################################################################
#
# Example:
# ~~~~~~~~
# Let f(x)=x*exp(x). This function satisfies the first-order differential
# equation
#
# xf’(x)-(x+1)f(x)=0.
#
# We are loking for the recurrence equation for a_k[f]. We should initialize
# Maple session as follows:
#
# >read(‘jacobi.mpl‘):
# >Jacobi([-x-1,x]);
#
###################################################################

with(LREtools):

bD:=f->D(unapply(f,x))(x):
bV:=f->(x-zeta)*bD(f)+ni(zeta)*f:
bK:=f->bV(bD(f)):
bU:=f->sigma(x)*bD(f)+tau(x)*f:
bL:=f->bU(bD(f)):
zeta:=1: zeta2:=-1:
_gamma:=alpha+beta+1: _delta:=alpha-beta:
ni:=z->if z=1 then alpha+1 else beta+1 fi:
mi:=(q,r,z)->pochhammer(k+ni(z),q+r)/(product(delta0(k+v),v=1..q)):
sigma:=x->x^2-1: tau:=x->(_gamma+1)*x+_delta:
h:=k->2^(2*k+_gamma)*k!*GAMMA(k+alpha+1)*GAMMA(k+beta+1)/
      (GAMMA(2*k+_gamma+1)*pochhammer(k+_gamma,k)):
lambda:=k->-k*(k+_gamma):
ksi0:=k->4*k*(k+alpha)*(k+beta)*(k+_gamma-1)/(pochhammer(2*k+_gamma-2,3)*
         (2*k+_gamma-1)):
ksi1:=k->-_delta*(_gamma-1)/((2*k+_gamma-1)*(2*k+_gamma+1)):
ksi2:=k->1:
delta0:=k->4*(k+alpha)*(k+beta)*pochhammer(k+_gamma-1,2)/
             (pochhammer(2*k+_gamma-2,3)*(k+_gamma)*(2*k+_gamma-1)):
delta1:=k->2*_delta/((2*k+_gamma-1)*(2*k+_gamma+1)):
delta2:=k->-1/(k+_gamma):
pX:=f->ksi0(k)*shift(f,k,-1)+ksi1(k)*f+ksi2(k)*shift(f,k,1):
pD:=f->delta0(k)*shift(f,k,-1)+delta1(k)*f+delta2(k)*shift(f,k,1):
lam_pI:=f->-lambda(k)*f:
lambda_pD:=f->-lambda(k)*pD(f):
pP:=f->f+zeta/2*pochhammer(2*k+_gamma,2)*pochhammer(2*k+_gamma+1,2)/
        ((k+ni(zeta2))*(k+_gamma)*pochhammer(2*k+_gamma+m,2))*shift(f,k,1):
pR:=f->delta0(k)*shift(f,k,-1)-2*zeta*(k+ni(zeta2)+m-1)/
                              pochhammer(2*k+_gamma+m-1,2)*f:
pS:=proc(z,i,j,f)
  global m,zeta,zeta2;
  local l,S;
   S:=(pP@@0)(f);
    if z=1 then zeta:=z; zeta2:=-1
      else zeta:=z; zeta2:=1 fi;
    for l from j to i

     do
       m:=l;
       S:=pP(S)
     od;
   S
end:

pZ:=proc(z,n,f)
   pS(z,n,1,f)
end:

pM:=proc(z,i,j,f)
  global m,zeta,zeta2;
  local l,R;
   R:=(pR@@0)(f);
    if z=1 then zeta:=z; zeta2:=-1
      else zeta:=z; zeta2:=1 fi;
    for l from j to i by -1
     do
       m:=l;
       R:=pR(R)
     od;
   R
end:

pN:=proc(z,n,f)
   pM(z,1,n,f)
end:

find_bQi:=proc(i,w)
  local wx,a,b,d,p,r,s,m,omega,eps;
    if simplify(w)=0 then
      RETURN([0,0,0,0,0,0,0])
    fi;
   wx:=unapply(simplify(w),x);
   d:=degree(wx(x),x);
   a:=degree(gcd((x-1)^d,wx(x)),x);
   b:=degree(gcd((x+1)^d,wx(x)),x);
   p:=simplify(wx(x)/((x-1)^a*(x+1)^b));
   r:=min(a,b); s:=max(a,b)-r; m:=floor((i+1)/2); omega:=i mod 2;
   if r=a then eps:=-1 else eps:=1 fi;
   if r>=m then
       [0,0,eps,0,m-omega,omega,simplify(p*(x^2-1)^(r-m)*(x-eps)^s)]
     elif (r+s>=m and m>r) then
       [0,m-r-omega,eps,omega,r,0,simplify(p*(x-eps)^(r+s-m))]
     else
       [i-2*r-2*s,s,eps,0,r,0,p]
    fi
end:

canonical_form:=proc(n,p)
  global zeta;
  local i,j,k,w,bQ,q,P;
   P[n]:=p;
    for k from 0 to n
     do
       w[k]:=coeff(P[n],(bD@@k)(f(x)),1)
     od;
   bQ:=array(0..n);
    for i from n to 1 by -1

24 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

     do
       q:=find_bQi(i,w[i]);
       bQ[i]:=q;
       zeta:=q[3];
       P[i-1]:=P[i]-(bD@@q[1]@bK@@q[2]@bV@@q[4]@bL@@q[5]@bU@@q[6])(q[7]*f(x));
        for j from 0 to i-1
         do
           w[j]:=coeff(P[i-1],(bD@@j)(f(x)),1)
         od
     od;
    bQ[0]:=[0,0,0,0,0,0,factor(simplify(w[0]))];
    bQ
end:

poly_to_oper:=proc(w,f)
  local j,i,p;
    for i from 0 to degree(w,x)
     do
       p[i]:=coeff(w,x,i)*(pX@@i)(f)
     od;
   sum(p[j],j=0..degree(w,x))
end:

integral_canonical_form:=proc(n,Q)
  local Omega,z,z2,t,i,j,l,p,p2,q,q2,e,epsilon,epsilon2,
        pom,O,sO,oO,stp,stm,sm,sp;
   z:=1; z2:=-1;
   Omega[z]:=[]; Omega[z2]:=[]; stp:=0; stm:=0;
   p[z]:=0; p[z2]:=0; q[z]:=0; q[z2]:=0;
    for i from 0 to n
     do
        if is(Q[i][3]=z) then
          Omega[z]:=[op(Omega[z]),i];
          p[z]:=max(p[z],Q[i][1]);
          q[z]:=max(q[z],Q[i][1]+2*Q[i][2]+Q[i][4]);
        else
          Omega[z2]:=[op(Omega[z2]),i];
          p[z2]:=max(p[z2],Q[i][1]);
          q[z2]:=max(q[z2],Q[i][1]+2*Q[i][2]+Q[i][4])
        fi
     od;
   q[z]:=q[z]-p[z];
   q[z2]:=q[z2]-p[z2];
   if is(p[z]+q[z]<p[z2]+q[z2]) then epsilon:=z2; epsilon2:=z
     else epsilon:=z; epsilon2:=z2 fi;
   p2:=max(p[epsilon],p[epsilon2]+q[epsilon2]);
   q2:=p[epsilon]+q[epsilon]-p2;
    for e in [epsilon,epsilon2]
     do
      for j in Omega[e]
       do
         pom:=(lam_pI@@Q[j][5]@lambda_pD@@Q[j][6])(poly_to_oper(Q[j][7],b[k]));
         pom:=mi(Q[j][2],Q[j][4],e)*shift(pZ(-e,Q[j][4],pom),k,Q[j][2]);
         pom:=pM(e,q[e]+1,q[e]+p[e]-Q[j][1],
                 pS(e,q[e]+p[e]-Q[j][1],2*Q[j][2]+Q[j][4]+1,pom));
         sp:=degree(Q[j][7],x)+Q[j][6]-Q[j][2]-Q[j][1]+q[e]+p[e];
         sm:=degree(Q[j][7],x)+Q[j][6]+Q[j][4]-Q[j][2]-Q[j][1]+p[e];
         if is(e=epsilon) then
            O[j]:=pM(e,q2+1,q[e],pom);
            sm:=sm+q[e]-q2

           else
            O[j]:=pZ(epsilon,q2,pN(e,p2-p[e],
                     pS(e,p2-p[e],q[e]+1,pom)));
            sp:=sp+p2+q2-p[e]-q[e];
            sm:=sm+p2-p[e]
          fi;
        stp:=max(stp,sp);
        stm:=max(stm,sm)
       od
     od;
   sO:=0;
   oO:=sum(O[t],t=0..n);
    for l from -stm to stp
      do
        sO:=sO+factor(simplify(coeff(oO,b[k+l],1)))*b[k+l]
      od;
   [sO,stm,stp,[epsilon,q2,p2]]
end:

Jacobi:=proc(lp)
  options ‘Copyright 2001 by Pawel Wozny‘;
  local Pp,np,Kp,m,p,i,j,l,oper,d,d2,A,B,pom,sp,sm,prze,w,o1,o2,o3;
   np:=nops(lp);
   if np=0 then Pp:=0 else
      np:=np-1; Pp:=sum(simplify(lp[l+1])*(bD@@l)(f(x)),l=0..np)
    fi;
   Kp:=canonical_form(np,Pp);
   oper:=integral_canonical_form(np,Kp);
   sm:=0; sp:=0; d:=1; d2:=0;
   for i from -oper[2] to oper[3]
     do
       pom:=coeff(oper[1],b[k+i],1);
       B[i]:=factor(simplify(pom*h(k+i)/h(k)));
       if B[i]<>0 then
          d:=lcm(d,denom(B[i]));
          d2:=gcd(d2,numer(B[i]));
          sp:=max(sp,i);
          sm:=min(sm,i)
        fi
     od;
   prze:=-sm-floor((sp-sm)/2);
   w:=factor(simplify(d/d2));
   o1:=oper[4][1]; o2:=oper[4][2]; o3:=oper[4][3];
   for j from sm+prze to sp+prze
     do
       A[j]:=0;
       if B[j-prze]<>0 then
          A[j]:=factor(simplify(shift(B[j-prze]*w,k,prze)));
          print(‘A[‘.j.‘]‘=A[j])
        fi
     od;
   print(’w’=shift(w,k,prze));
   print(’s’=prze,’omega’=o1,’d’=o2,’e’=o3);
   print(T=1/’h’(k+prze)*E^’s’*Z[’omega’]^‘(d)‘*D^’e’);
   [sm+prze,sp+prze,A,shift(w,k,prze)]
end:

B.2. Expansions in the Hahn polynomials.

# Constructing recurrence equation for Hahn series coefficients
# of a solution of a linear difference equation.

26 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

#
###################################################################
#
# Version 1.0 for MapleV release 4
#
# Copyright (c) Pawel Wozny (Pawel.Wozny@ii.uni.wroc.pl), 07/02/2001
#
###################################################################
#
# Let f(x) be a solution of a linear difference equation
#
# n m
# ----- -----
# \ i \ j
# (1) ) u_i(x) (D-) f(x) + ) v_j(x) (D+) f(x) = g(x)
# / /
# ----- -----
# i = 0 j = 0
#
# where u_i(x), v_j(x) are polynomials and
#
# (D+)(f(x)):=f(x+1)-f(x), (D-)(f(x)):=f(x)-f(x-1).
#
# Let a_k[f] be the Fourier coefficients of the function f(x) in the
# expansion
# -----
# \
# (2) f(x) = ) a_k[f] Q_k(x),
# /
# -----
# k
#
# where Q_k(x) is the k-th monic Hahn polynomial. This program finds
# recurrence relation for a_k[f] of the form
#
# r + d
# -----
# \
# (3) ) A_i(k) a_{i+r}[f] = B(k)
# /
# -----
# i = -r
#
# under assumption that a_k[g] are known explicitly. Here r is a natural
# number and d {0,1}.
#
# Main procedure
# ~~~~~~~~~~~~~~
# - Hahn(U,V)
#
# Input
# ~~~~~
# - vectors U := [u_0(x),u_1(x),...,u_n(x)] and
# V := [v_0(x),v_1(x),...,v_m(x)]
#
# Output
# ~~~~~~
# - vector [-r,r+d,A,w], where A[-r..r+d] is a table of A_l(k)
# (l=-r..r+d)
# - operator T such that B(k)= w*T(a_k[g])

# (see Algorithm 3.1, Step 6, Eq. (**)) is given on the screen
#
###################################################################
#
# Example:
# ~~~~~~~~
# Let f(x)=product(x+i,i=0..n-1) (Pochhammer’s polynomial).
# This function satisfies the first-order
# difference equation
#
# (x+n-1)(D+)(f(x))-nf(x)=0.
#
# We are loking for the recurrence equation for a_k[f]. We should initialize
# Maple session as follows:
#
# >read(‘hahn.mpl‘):
# >Hahn([],[-n,x+n-1]);
#
###################################################################

with(LREtools):
sigma:=x->if R=‘+‘ then -(x+beta+1)*(x-N+1) else x*(N+alpha-x) fi:
tau:=x->(beta+1)*(N-1)-(alpha+beta+2)*x:
bD:=f->if R=‘+‘ then delta(f,x) else f-shift(f,x,-1) fi:
bU:=f->sigma(x)*bD(f,x)+tau(x)*f:
bV:=f->if R=‘+‘ then (x-zeta_p)*bD(f,x)+(N+alpha-zeta_p)*f
        else (x-zeta_m)*bD(f)+(beta+1+zeta_m)*f fi:
zeta_p:=N-1: zeta_p2:=-beta-1: zeta_m:=0: zeta_m2:=N+alpha:
_gamma:=alpha+beta+1:
h:=k->k!*GAMMA(k+alpha+1)*GAMMA(k+beta+1)*pochhammer(2*k+_gamma+1,N-k-1)/
      (pochhammer(k+_gamma,k)*(N-k-1)!):
lambda:=k->k*(k+_gamma):
ksi0:=k->k*(N-k)*(k+alpha)*(k+beta)*(k+_gamma-1)*(k+_gamma+N-1)/
         (pochhammer(2*k+_gamma-2,2)*pochhammer(2*k+_gamma-1,2)):
ksi1:=k->((alpha-beta+2*N-2)/4+(beta^2-alpha^2)*(_gamma+2*N-1)/
                              (4*(2*k+_gamma-1)*(2*k+_gamma+1))):
ksi2:=k->1:
delta_p0:=k->(N-k)*(k+alpha)*(k+beta)*pochhammer(k+_gamma-1,2)*(k+_gamma+N-1)
             /(pochhammer(2*k+_gamma-2,2)*pochhammer(2*k+_gamma-1,2)*(k+_gamma)):
delta_p1:=k->-(2*k*(k+_gamma)+(_gamma-alpha)*(_gamma-1)-
              N*(alpha-beta))/((2*k+_gamma-1)*(2*k+_gamma+1)):
delta_p2:=k->-1/(k+_gamma):
delta_m0:=k->delta_p0(k):
delta_m1:=k->1+delta_p1(k):
delta_m2:=k->delta_p2(k):
pX:=f->ksi0(k)*shift(f,k,-1)+ksi1(k)*f+ksi2(k)*shift(f,k,1):
pD:=f->if R=‘+‘ then
         delta_p0(k)*shift(f,k,-1)+delta_p1(k)*f+delta_p2(k)*shift(f,k,1):
        else
         delta_m0(k)*shift(f,k,-1)+delta_m1(k)*f+delta_m2(k)*shift(f,k,1):
       fi:
lambda_pD:=f->if R=‘+‘ then
                -lambda(k)*(delta_m0(k)*shift(f,k,-1)+
                 delta_m1(k)*f+delta_m2(k)*shift(f,k,1))
               else
                -lambda(k)*(delta_p0(k)*shift(f,k,-1)+
                 delta_p1(k)*f+delta_p2(k)*shift(f,k,1))
              fi:
pP:=f->if R=‘+‘ then
         f+pochhammer(2*k+_gamma,2)*pochhammer(2*k+_gamma+1,2)/

28 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

         (pochhammer(2*k+_gamma+m,2)*(k+_gamma)*(k+_gamma+zeta_p+1)*
         (k-zeta_p2))*shift(f,k,1)
        else
         f-pochhammer(2*k+_gamma,2)*pochhammer(2*k+_gamma+1,2)/
          (pochhammer(2*k+_gamma+m,2)*(k+_gamma)*(k+_gamma+N-zeta_m)*
          (k-N+zeta_m2+1))*shift(f,k,1);
       fi:
pQ:=f->if R=‘+‘ then
         (k+N+alpha-zeta_p)*f-
         pochhammer(2*k+_gamma,2)*pochhammer(2*k+_gamma+1,2)*
         (k+m-zeta_p2-1)/(pochhammer(2*k+_gamma+m,2)*(k+_gamma)*
         (k+_gamma+zeta_p+1)*(k-zeta_p2))*shift(f,k,1)
        else
         (k+beta+zeta_m+1)*f+
         pochhammer(2*k+_gamma,2)*pochhammer(2*k+_gamma+1,2)*(k+m-N+zeta_m2)/
         (pochhammer(2*k+_gamma+m,2)*(k+_gamma)*(k+_gamma+N-zeta_m)*
         (k-N+zeta_m2+1))*shift(f,k,1)
       fi:
pR:=f->if R=‘+‘ then
         delta_p0(k)*shift(f,k,-1)-
         (k+m+_gamma+zeta_p)*(k+m-zeta_p2-1)/pochhammer(2*k+_gamma+m-1,2)*f
        else
         delta_m0(k)*shift(f,k,-1)+
         (k+m+_gamma+N-zeta_m-1)*(k+m-N+zeta_m2)/
         pochhammer(2*k+_gamma+m-1,2)*f
        fi:
pF:=f->if R=‘+‘ then
         delta_p0(k)/(k+zeta_m2-zeta_p-1)*shift(f,k,-1)+
         (k+m+_gamma+zeta_p-zeta_m)/pochhammer(2*k+_gamma+m-1,2)*f;
        else
         delta_m0(k)/(k-zeta_p2+zeta_m-1)*shift(f,k,-1)-
         (k+m+_gamma+zeta_p-zeta_m)/pochhammer(2*k+_gamma+m-1,2)*f
       fi:

find_bQi:=proc(i,w,r)
  local test1,test2,j,wx,q,a,b,z1,z2,s;
   if simplify(w)=0 then
      RETURN([0,0,0,0,0,r])
    fi;
   test1:=true; test2:=true; a:=0; b:=0;
   wx:=unapply(simplify(w),x);
   if r=‘+‘ then
      z1:=-beta-1; z2:=N-1; s:=-1
     else
      z1:=0; z2:=N+alpha; s:=1
    fi;
   for j from 1 to i while (test1 or test2)
     do
        if (test1 and simplify(wx(-s*j+z1+s*i))=0) then
          a:=a+1 else test1:=false fi;
        if (test2 and simplify(wx(-s*j+z2+s*i))=0) then
          b:=b+1 else test2:=false fi
     od;
   q:=simplify(wx(x)/(product(x-s*i-z1+s*k,k=1..a)*
                      product(x-s*i-z2+s*l,l=1..b)));
    if min(a,b)=b then
      [i-a,a-b,z1,b,(-s)^i*(-1)^b*factor(simplify(shift(q,x,s*i))),r]
     else
      [i-b,b-a,z2,a,(-s)^i*(-1)^a*factor(simplify(shift(q,x,s*i))),r]
    fi
end:

canonical_form:=proc(n,p,r)
  global zeta_p,zeta_m,R;
  local i,j,k,w,bQ,q,s,P;
   R:=r;
   P[n]:=p;
    if R=‘+‘ then
      s:=1 else s:=-1 fi;
    for k from 0 to n
     do
       w[k]:=coeff(P[n],f(x+s*k),1)
     od;
   bQ:=array(0..n);
    for i from n to 1 by -1
     do
       q:=find_bQi(i,w[i],r);
       bQ[i]:=q;
        if r=‘+‘ then zeta_p:=q[3] else zeta_m:=q[3] fi;
       P[i-1]:=P[i]-(bD@@q[1]@bV@@q[2]@bU@@q[4])(q[5]*f(x));
        for j from 0 to i-1
         do
           w[j]:=coeff(P[i-1],f(x+s*j),1)
         od
     od;
    bQ[0]:=[0,0,0,0,factor(simplify(w[0])),r];
    bQ
end:

poly_to_oper:=proc(w,f)
  local i,p,j;
    for i from 0 to degree(w,x)
     do
       p[i]:=coeff(w,x,i)*(pX@@i)(f)
     od;
   sum(p[j],j=0..degree(w,x))
end:

pS:=proc(z,i,j,r,f)
  global R,m,zeta_p,zeta_p2,zeta_m,zeta_m2;
  local l,S;
   R:=r;
   S:=(pP@@0)(f);
   if r=‘+‘ then
       if is(z=N-1) then zeta_p:=z; zeta_p2:=-beta-1
        else zeta_p:=z; zeta_p2:=N-1 fi
     else
      if is(z=0) then zeta_m:=z; zeta_m2:=N+alpha
       else zeta_m:=z; zeta_m2:=0 fi
    fi;
    for l from j to i
     do
       m:=l;
       S:=pP(S)
     od;
   S
end:

pZ:=proc(z,n,r,f)
   pS(z,n,1,r,f)

30 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

end:

pU:=proc(z,n,r,f)
  global R,m,zeta_p,zeta_p2,zeta_m,zeta_m2;
  local l,Q;
  R:=r;
  Q:=(pQ@@0)(f);
  if r=‘+‘ then
       if is(z=N-1) then zeta_p:=z; zeta_p2:=-beta-1
        else zeta_p:=z; zeta_p2:=N-1 fi
     else
      if is(z=0) then zeta_m:=z; zeta_m2:=N+alpha
       else zeta_m:=z; zeta_m2:=0 fi
    fi;
   for l from 1 to n
     do
       m:=l;
       Q:=pQ(Q)
     od;
   Q
end:

pM:=proc(z,i,j,r,f)
  global R,m,zeta_p,zeta_p2,zeta_m,zeta_m2;
  local l,_R;
   R:=r;
   _R:=(pR@@0)(f);
    if r=‘+‘ then
       if is(z=N-1) then zeta_p:=z; zeta_p2:=-beta-1
        else zeta_p:=z; zeta_p2:=N-1 fi
     else
      if is(z=0) then zeta_m:=z; zeta_m2:=N+alpha
       else zeta_m:=z; zeta_m2:=0 fi
    fi;
    for l from j to i by -1
     do
       m:=l;
       _R:=pR(_R)
     od;
   _R
end:

pN:=proc(z,n,r,f)
   pM(z,1,n,r,f)
end:

pW:=proc(zp,zm,i,j,r,f)
  global R,m,zeta_p,zeta_p2,zeta_m,zeta_m2;
  local l,V;
   R:=r;
   V:=(pF@@0)(f);
   if is(zp=N-1) then zeta_p:=zp; zeta_p2:=-beta-1
     else zeta_p:=zp; zeta_p2:=N-1 fi;
   if is(zm=0) then zeta_m:=zm; zeta_m2:=N+alpha
     else zeta_m:=zm; zeta_m2:=0 fi;
    for l from j to i by -1
     do
       m:=l;
       V:=pF(V)
     od;

   V
end:

pY:=proc(zp,zm,n,r,f)
   pW(zp,zm,1,n,r,f)
end:

integrate_canonical_form:=proc(n,Q,r)
  global R;
  local Omega,z,z2,i,j,t,l,p,p2,q,q2,e,epsilon,epsilon2,
        pom,O,sO,oO,stp,stm,sm,sp;
  R:=r;
  if r=‘+‘ then
      z:=N-1; z2:=-beta-1
     else
      z:=0; z2:=N+alpha
    fi;
  Omega[z]:=[]; Omega[z2]:=[]; stp:=0; stm:=0;
  p[z]:=0; p[z2]:=0; q[z]:=0; q[z2]:=0;
  for i from 0 to n
     do
        if is(Q[i][3]=z) then
          Omega[z]:=[op(Omega[z]),i];
          p[z]:=max(p[z],Q[i][1]);
          q[z]:=max(q[z],Q[i][1]+Q[i][2]);
        else
          Omega[z2]:=[op(Omega[z2]),i];
          p[z2]:=max(p[z2],Q[i][1]);
          q[z2]:=max(q[z2],Q[i][1]+Q[i][2])
        fi
     od;
   q[z]:=q[z]-p[z];
   q[z2]:=q[z2]-p[z2];
   if is(p[z]+q[z]<p[z2]+q[z2]) then epsilon:=z2; epsilon2:=z
     else epsilon:=z; epsilon2:=z2 fi;
   p2:=max(p[epsilon],p[epsilon2]+q[epsilon2]);
   q2:=p[epsilon]+q[epsilon]-p2;
   for e in [epsilon,epsilon2]
     do
      for j in Omega[e]
       do
         pom:=(lambda_pD@@Q[j][4])(poly_to_oper(Q[j][5],b[k]));
         pom:=pU(Q[j][3],Q[j][2],r,pom);
         pom:=pM(e,q[e]+1,q[e]+p[e]-Q[j][1],r,
                 pS(e,q[e]+p[e]-Q[j][1],Q[j][2]+1,r,pom));
         sp:=degree(Q[j][5],x)+Q[j][4]+q[e]+p[e]-Q[j][1];
         sm:=degree(Q[j][5],x)+Q[j][4]+p[e]-Q[j][1];
          if is(e=epsilon) then
            O[j]:=pM(e,q2+1,q[e],r,pom);
            sm:=sm+q[e]-q2
           else
            O[j]:=pZ(epsilon,q2,r,pN(e,p2-p[e],r,
                     pS(e,p2-p[e],q[e]+1,r,pom)));
            sp:=sp+p2+q2-p[e]-q[e];
            sm:=sm+p2-p[e]
          fi;
        stp:=max(stp,sp);
        stm:=max(stm,sm)
       od
     od;

32 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

   sO:=0;
   oO:=sum(O[t],t=0..n);
    for l from -stm to stp
      do
        sO:=sO+factor(simplify(coeff(oO,b[k+l],1)))*b[k+l]
      od;
   [epsilon,q2,p2,sO,stm,stp]
end:

p_i_m:=proc(zeta,i,r,pMp,stmp,stpp,eta,j,s,pMm,stmm,stpm)
  global R,m,zeta_p,zeta_p2,zeta_m,zeta_m2;
  local p,q,pom,pUp,pUm,z,sO,oO,stm,stp,l,op;
  if is(zeta=N-1) then zeta_p:=zeta; zeta_p2:=-beta-1
      else zeta_p:=zeta; zeta_p2:=N-1 fi;
  if is(eta=0) then zeta_m:=eta; zeta_m2:=N+alpha
      else zeta_m:=zeta; zeta_m2:=0 fi;
   p:=min(i,j); q:=max(i,j);
  if is(q=i) then
      op:=[‘+‘,zeta_p,q-p,p,zeta_p,zeta_m2,s,r];
      R:=‘-‘;
      m:=p+1;
      pom:=((pR@pP-pP@@0)@@r)(pMm);
      pom:=pY(zeta_p2,zeta_m,p,‘-‘,pom);
      z:=zeta_p;
      zeta_p:=zeta_p2;
      zeta_p2:=z;
      R:=‘+‘;
      pUm:=pZ(zeta_p,q-p,‘+‘,pom);
      m:=q+1;
      pom:=((pR@pP+pP@@0)@@s)(pMp);
      pUp:=pW(zeta_p,zeta_m2,q-p+1,q,‘+‘,pom);
      stm:=max(r+p+stmm,s+p+stmp);
      stp:=max(r+q-p+stpm,s+stpp)
     else
      op:=[‘-‘,zeta_m,q-p,p,zeta_p,zeta_m2,s,r];
      R:=‘+‘;
      m:=p+1;
      pom:=((pR@pP+pP@@0)@@s)(pMp);
      pom:=pY(zeta_p,zeta_m2,p,‘+‘,pom);
      z:=zeta_m;
      zeta_m:=zeta_m2;
      zeta_m2:=z;
      R:=‘-‘;
      pUp:=pZ(zeta_m,q-p,‘-‘,pom);
      m:=q+1;
      pom:=((pR@pP-pP@@0)@@r)(pMm);
      pUm:=pW(zeta_p2,zeta_m,q-p+1,q,‘-‘,pom);
      stm:=max(r+p+stmm,s+p+stmp);
      stp:=max(s+q-p+stpp,r+stpm)
    fi;
   sO:=0;
   oO:=pUp+pUm;
    for l from -stm to stp
      do
        sO:=sO+factor(simplify(coeff(oO,b[k+l],1)))*b[k+l]
      od;
   [sO,stm,stp,op]
end:

Hahn:=proc(lm,lp)

  options ‘Copyright 2001 by Pawel Wozny‘;
  global R;
  local Pp,Pm,np,nm,Kp,Km,m,p,i,j,n,l,oper,w,d,d2,A,B,pom,sp,sm,prze,o;
   nm:=nops(lm); np:=nops(lp);
   R:=‘-‘;
    if nm=0 then Pm:=0 else
      nm:=nm-1; Pm:=sum(simplify(lm[n+1])*(bD@@n)(f(x)),n=0..nm)
    fi;
   R:=‘+‘;
    if np=0 then Pp:=0 else
      np:=np-1; Pp:=sum(simplify(lp[l+1])*(bD@@l)(f(x)),l=0..np)
    fi;
   Km:=canonical_form(nm,Pm,‘-‘);
   Kp:=canonical_form(np,Pp,‘+‘);
   m:=integrate_canonical_form(nm,Km,‘-‘);
   p:=integrate_canonical_form(np,Kp,‘+‘);
   oper:=p_i_m(p[1],p[2],p[3],p[4],p[5],p[6],m[1],m[2],m[3],m[4],m[5],m[6]);
   sm:=0; sp:=0; d:=1; d2:=1;
    for i from -oper[2] to oper[3]
     do
       pom:=coeff(oper[1],b[k+i],1);
       B[i]:=factor(simplify(pom*h(k+i)/h(k)));
        if B[i]<>0 then
          d:=lcm(d,denom(B[i]));
          d2:=gcd(d2,numer(B[i]));
          sp:=max(sp,i);
          sm:=min(sm,i)
        fi
     od;
   prze:=-sm-floor((sp-sm)/2);
   w:=factor(simplify(d/d2));
   o:=oper[4];
    for j from sm+prze to sp+prze
     do
       A[j]:=0;
        if B[j-prze]<>0 then
          A[j]:=factor(simplify(shift(B[j-prze]*w,k,prze)));
          print(‘A[‘.j.‘]‘=A[j])
        fi
     od;
   print(’w’=shift(w,k,prze));
    if o[1]=‘+‘ then
      print(’s’=prze,’zeta’=o[2],’eta’=o[6],’d’=o[3],’t’=o[4],
            ’u’=o[7],’v’=o[8]);
      print(T=1/’h’(k+prze)*E^’s’*Z[’zeta’]^‘+(d)‘*Y[’zeta’,’eta’]^‘+(t)‘*
            Z[’zeta’]^‘+(t)‘*(D^‘-‘)^’u’*(D^‘+‘)^’v’)
     else
      print(’s’=prze,’epsilon’=o[2],’zeta’=o[5],’eta’=o[6],’d’=o[3],
            ’t’=o[4],’u’=o[7],’v’=o[8]);
      print(T=1/’h’(k+prze)*E^’s’*Z[’epsilon’]^‘-(d)‘*
            Y[’zeta’,’eta’]^‘+(t)‘*Z[’zeta’]^‘+(t)‘*(D^‘-‘)^’u’*(D^‘+‘)^’v’)
    fi:
   [sm+prze,sp+prze,A,shift(w,k,prze)]
end:

B.3. Expansions in the little q-Jacobi polynomials.

###################################################################
#
# Warning!!! To run this program you need file qsum.mpl, which is available
# ~~~~~~~~~~ on http://www.imn.htwk-leipzig.de/~koepf.

34 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

#
###################################################################
#
# Constructing recurrence equation for the little q-Jacobi series
# coefficients of a solution of a linear q-difference equation.
#
###################################################################
#
# Version 1.0 for MapleV release 4
#
# Copyright (c) Pawel Wozny (Pawel.Wozny@ii.uni.wroc.pl), 07/02/2001
#
###################################################################
#
# Let f(x) be a solution of a linear q-difference equation
#
# n m
# ----- -----
# \ i \ j
# (1) ) u_i(x) (D-) f(x) + ) v_j(x) (D+) f(x) = g(x),
# / /
# ----- -----
# i = 0 j = 0
#
# where u_i(x), v_j(x) are polynomials, and
#
# f(q x) - f(x)
# (D+) = -------------,
# x (q - 1)
#
# f(x/q) - f(x)
# (D-) = ------------- (0 < q <1 ).
# x (1/q - 1)
#
# Let a_k[f] be the Fourier coefficients of the function f(x),
#
# -----
# \
# (2) f(x) = ) a_k[f] p_k(x),
# /
# -----
# k
#
# where p_k(x) is the k-th monic little q-Jacobi polynomial. This program
# finds recurrence relation for a_k[f], of the form
#
# r + d
# -----
# \
# (3) ) A_i(k) a_{i+r}[f] = B(k)
# /
# -----
# i = -r
#
# under assumption that a_k[g] are known explicitly. Here r is a natural
# number and d {0,1}.
#
# Main procedure
# ~~~~~~~~~~~~~~
# - l_qJacobi(U,V)

#
# Input
# ~~~~~
# - vectors U := [u_0(x),u_1(x),...,u_n(x)] and
# V := [v_0(x),v_1(x),...,v_m(x)]
#
# Output
# ~~~~~~
# - vector [-r,r+d,A], where A[-r..r+d] is a table of A_l(k)
# (l=-r..r+d)
# - operator T such that B(k)= T(a_k[g])
# (see Algorithm 3.1, Step 6, Eq. (**)) is given on the screen
#
###################################################################
#
# Example:
# ~~~~~~~~
# Let f(x)=x^n. This function satisfies the first-order
# q-difference equation
#
# x(D+)(f(x))-[(q^n-1)/(q-1)]f(x)=0.
#
# We are loking for the recurrence equation for a_k[f]. We should initialize
# Maple session as follows:
#
# >read(‘l_jacobi.mpl‘):
# >l_qJacobi([],[-qnum(n),x]);
#
###################################################################

read(‘qsum.mpl‘):
with(LREtools):
bE:=(f,i)->subs({x=q^i*x,X=q^i*X},f):
qpochhammer:=(x,q,kk)->product(1-x*q^mm,mm=0..kk-1):
qnum:=kk->(q^kk-1)/(q-1):
eps:=()->if R=‘+‘ then 1 else q fi:
ww:=(n,q)->if (is(n=0) or is(q=1)) then 1
            else
             if R=‘+‘ then 1/(q^((n+1)*(n+2)/2-2*n-1)*(q-1)^n)
               else 1/(q^(-(n+1)*(n+2)/2+2*n+1)*(q^(-1)-1)^n)
             fi
           fi:
bD:=f->if R=‘+‘ then (bE(f,1)-f)/(X*(q-1)) else (bE(f,-1)-f)/(X*(q^(-1)-1)) fi:
bU:=f->sigma(x)/eps()*bD(f,x)+tau(x)*f:
bV:=f->if R=‘+‘ then (x-zeta_p)*bD(f,x) else (x-zeta_m)/q*bD(f) fi:
zeta_p2:=0: zeta_p:=1/(b*q): zeta_m2:=1: zeta_m:=0:
ni_p:=(zp2,zm2)->qsimplify(2*tau(zm2)/(2*a*b*q^2*(zm2-zp2))):
ni_m:=(zp2,zm2)->qsimplify(2*tau(zp2)/(2*(zp2-zm2))):
gamma_p:=(zp2,zm2,k)->qsimplify(-q^(k-1)*lambda(k)/RR(k-1)+ni_p(zp2,zm2)):
gamma_m:=(zp2,zm2,k)->qsimplify(1/(q-1-q^(1-k)*RR(k-1)/lambda(k))+ni_m(zp2,zm2)):
sigma:=x->if R=‘+‘ then a*q*x*(b*q*x-1) else x*(x-1) fi:
sigmam:=x->x*(x-1):
tau:=x->(a*q*(b*q*x-1)+1-x)/(q-1):
lambda:=k->-(q^k-1)*(a*b*q^(k+1)-1)/(q^(k-1)*(q-1)^2):
v_p:=k->qsimplify(-q^(k-1)*lambda(k)/RR(k-1)):
v_m:=k->qsimplify(1/(q-1-q^(1-k)*RR(k-1)/lambda(k))):
t_1:=diff(tau(x),x): t_0:=tau(0):
s_2:=simplify(diff(sigmam(x),x,x)/2+(q-1)*t_1):
s_1:=simplify(subs(x=0,diff(sigmam(x),x))+(q-1)*tau(0)):
s_0:=sigmam(0):

36 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

RR:=k->s_2*qnum(k)+t_1:
h:=n->if n=0 then 1 elif n>0 then product(ksi0(k+ii),ii=1..n)
         else product(1/ksi0(k+ii),ii=n+1..0) fi:
ksi0:=k->a*q^(2*k)*(q^k-1)*(a*q^k-1)*(b*q^k-1)*(a*b*q^k-1)
          /((a*b*q^(2*k)-1)^2*(a*b*q^(2*k)-q)*(a*b*q^(2*k+1)-1)):
ksi1:=k->q^k*(1+a+a*(1+a)*b*q^(2*k+1)-a*(1+b)*q^k*(1+q))
          /((a*b*q^(2*k)-1)*(a*b*q^(2*k+2)-1)):
ksi2:=k->1:
delta_p0:=k->qsimplify(-q^(1-k)*RR(k-1)*ksi0(k)/lambda(k)):
delta_p1:=k->qsimplify(-q*t_0/((q+1)*lambda(k))-
                      q*ksi1(k)*((q-1)*qnum(-k)*RR(k-1)+t_1)/
                      ((q+1)*lambda(k))):
delta_p2:=k->qsimplify(s_2*qnum(k)/lambda(k)):
delta_m0:=k->qsimplify(delta_p0(k)+(q-1)*ksi0(k)):
delta_m1:=k->qsimplify(delta_p1(k)+(q-1)*ksi1(k)):
delta_m2:=k->qsimplify(delta_p2(k)+(q-1)*ksi2(k)):
pX:=f->ksi0(k)*shift(f,k,-1)+ksi1(k)*f+ksi2(k)*shift(f,k,1):
pD:=f->if R=‘+‘ then
         delta_p0(k)*shift(f,k,-1)+delta_p1(k)*f+delta_p2(k)*shift(f,k,1):
        else
         delta_m0(k)*shift(f,k,-1)+delta_m1(k)*f+delta_m2(k)*shift(f,k,1):
       fi:
lambda_pD:=f->if R=‘+‘ then
                -lambda(k)/q*(delta_m0(k)*shift(f,k,-1)+
                 delta_m1(k)*f+delta_m2(k)*shift(f,k,1))
               else
                -lambda(k)*(delta_p0(k)*shift(f,k,-1)+
                 delta_p1(k)*f+delta_p2(k)*shift(f,k,1))
              fi:
pDelta:=f->if R=‘+‘ then
             if m>0 then (pP@pR)(f)+(q-1)*(pQ@pR)(f)+zeta_p*(q-1)*f else
               delta_m0(k)*shift(f,k,-1)+delta_m1(k)*f+
               delta_m2(k)*shift(f,k,1) fi
            else
             if m>0 then (pP@pR)(f)-(q-1)*(pQ@pR)(f)-zeta_m*(q-1)*f else
               delta_p0(k)*shift(f,k,-1)+delta_p1(k)*f+
               delta_p2(k)*shift(f,k,1) fi
            fi:
pP:=f->if R=‘+‘ then
         f+
         b*q^m*qpochhammer(a*b*q^(2*k+1),q,3)*(a*b*q^(2*k+2)-1)/
            ((b*q^(k+1)-b*q*zeta_p2)*(1-zeta_p*a*b*q^(k+2))*
            (a*b*q^(k+1)-1)*qpochhammer(a*b*q^(2*k+m+1),q,2))*shift(f,k,1)
        else
         f+
         q^(m-1)*qpochhammer(a*b*q^(2*k+1),q,3)*(a*b*q^(2*k+2)-1)
            /(q^k*(a*q^(k+1)-zeta_m)*
            (zeta_m2*b*q^(k+1)-1)*(a*b*q^(k+1)-1)*
            qpochhammer(a*b*q^(2*k+m+1),q,2))*shift(f,k,1)
       fi:
pQ:=f->if R=‘+‘ then
         v_p(k)*f+
         qpochhammer(a*b*q^(2*k+1),q,3)*(a*b*q^(2*k+2)-1)*(1-a*b*q^(k+m+1))/
            (a*q^(k+1)*(q-1)*(b*q^(k+1)-b*q*zeta_p2)*(1-zeta_p*a*b*q^(k+2))*
            (a*b*q^(k+1)-1)*qpochhammer(a*b*q^(2*k+m+1),q,2))*shift(f,k,1)
        else
         v_m(k)*f+
         q^(m-1)*qpochhammer(a*b*q^(2*k+1),q,3)*(a*b*q^(2*k+2)-1)*
            (1-a*b*q^(k+m+1))/(q^k*(q-1)*(a*q^(k+1)-zeta_m)*
            (zeta_m2*b*q^(k+1)-1)*(a*b*q^(k+1)-1)*

            qpochhammer(a*b*q^(2*k+m+1),q,2))*shift(f,k,1)
       fi:
pR:=f->if R=‘+‘ then
         delta_p0(k)*shift(f,k,-1)+
         a*q^k*(q-1)*(b*q^(k+m)-b*q*zeta_p2)*(1-zeta_p*a*b*q^(k+m+1))/
            qpochhammer(a*b*q^(2*k+m),q,2)*f
        else
         delta_m0(k)*shift(f,k,-1)+
         q^k*(q-1)*(a*q^(k+m)-zeta_m)*(zeta_m2*b*q^(k+m)-1)/
            qpochhammer(a*b*q^(2*k+m),q,2)*f
       fi:
pF:=f->if R=‘+‘ then
         ni_p(zeta_p2,zeta_m2)*delta_m0(k)/gamma_p(zeta_p2,zeta_m2,k-1)*
         shift(f,k,-1)+
         b*q^(k+1)*(q-1)*(a*q-zeta_m)*(1-zeta_p)*
            (zeta_m2*q^(k+m-1)-zeta_p2)/qpochhammer(a*b*q^(2*k+m),q,2)*f
        else
         q*ni_m(zeta_p2,zeta_m2)*delta_p0(k)/gamma_m(zeta_p2,zeta_m2,k-1)*
         shift(f,k,-1)
         -a*q^(2*k+m)*(b*q-zeta_m)*(q-1)*
            (zeta_m2*q^(k+m-1)-zeta_p2+zeta_m*(1+zeta_p2))*(a*b*q^2*zeta_p-1)/
            qpochhammer(a*b*q^(2*k+m),q,2)*f
       fi:

find_bQi:=proc(i,w,r)
  global R;
  local test1,test2,j,wx,wx2,wx3,qq,stala,aa,bb,z1,z2,s,m;
   R:=r;
    if qsimplify(w)=0 then
      RETURN([0,0,0,0,0,r])
    fi;
   test1:=true; test2:=true; aa:=0; bb:=0;
   wx:=unapply(qsimplify(w),x);
   wx2:=unapply(qsimplify(w),x);
   wx3:=unapply(qsimplify(w),x);
    if r=‘+‘ then
      z1:=0; z2:=1/(b*q); s:=-1; m:=a*b*q^2/eps()
     else
      z1:=1; z2:=0; s:=1; m:=1/eps()
    fi;
    for j from 1 to i while (test1 or test2)
     do
        if (test1 and simplify(wx2(z1*q^(s*(i-j))))=0) then
          aa:=aa+1; wx2:=unapply(qsimplify(wx2(x)/(x*q^(-s*(i-j))-z1)),x)
           else test1:=false fi;

        if (test2 and simplify(wx3(z2*q^(s*(i-j))))=0) then
          bb:=bb+1; wx3:=unapply(qsimplify(wx3(x)/(x*q^(-s*(i-j))-z2)),x)
           else test2:=false fi
     od;
   qq:=simplify(wx(x)/(product(x*q^(s*(k-i))-z1,k=1..aa)*
                      product(x*q^(s*(l-i))-z2,l=1..bb)));
    if min(aa,bb)=bb then
      stala:=ww(i-aa,q)*ww(aa-bb,q)*ww(bb,q)*q^(s*((i-aa)*(aa-bb)+bb*(i-bb)));
      [i-aa,aa-bb,z1,bb,factor(qsimplify(eps()^(aa-bb)/(stala*m^bb)*
       subs(X=x,bE(qq*X^i,s*i)))),r]
     else
      stala:=ww(i-bb,q)*ww(bb-aa,q)*ww(aa,q)*q^(s*((i-bb)*(bb-aa)+aa*(i-aa)));
      [i-bb,bb-aa,z2,aa,factor(qsimplify(eps()^(bb-aa)/(stala*m^aa)*
        subs(X=x,bE(qq*X^i,s*i)))),r]

38 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

    fi
end:

canonical_form:=proc(n,p,r)
  global zeta_p,zeta_m,R;
  local i,j,k,w,bQ,qq,s,P;
   R:=r;
   P[n]:=p;
    if R=‘+‘ then
      s:=1 else s:=-1 fi;
    for k from 0 to n
     do
       w[k]:=coeff(P[n],f(x*q^(s*k)),1)
     od;
   bQ:=array(0..n);
    for i from n to 1 by -1
     do
       qq:=find_bQi(i,w[i],r);
       bQ[i]:=qq;
        if r=‘+‘ then zeta_p:=qq[3] else zeta_m:=qq[3] fi;
       P[i-1]:=P[i]-(bD@@qq[1]@bV@@qq[2]@bU@@qq[4])(qq[5]*f(x));
        for j from 0 to i-1
         do
           w[j]:=coeff(P[i-1],f(x*q^(s*j)),1)
         od
     od;
    bQ[0]:=[0,0,0,0,factor(qsimplify(subs(X=x,w[0]))),r];
    bQ
end:

poly_to_oper:=proc(w,f)
  local i,p,j;
    for i from 0 to degree(w,x)
     do
       p[i]:=coeff(w,x,i)*(pX@@i)(f)
     od;
   sum(p[j],j=0..degree(w,x))
end:

pS:=proc(z,i,j,r,f)
  global R,m,zeta_p,zeta_p2,zeta_m,zeta_m2;
  local l,S;
   R:=r;
   S:=(pP@@0)(f);
    if r=‘+‘ then
       if is(z=0) then zeta_p:=z; zeta_p2:=1/(b*q)
        else zeta_p:=z; zeta_p2:=0 fi
     else
      if is(z=1) then zeta_m:=z; zeta_m2:=0
       else zeta_m:=z; zeta_m2:=1 fi
    fi;
    for l from j to i
     do
       m:=l;
       S:=pP(S)
     od;
   S
end:

pZ:=proc(z,n,r,f)

   pS(z,n,1,r,f)
end:

pU:=proc(z,n,r,f)
  global R,m,zeta_p,zeta_p2,zeta_m,zeta_m2;
  local l,Q;
   R:=r;
   Q:=(pQ@@0)(f);
    if r=‘+‘ then
       if is(z=0) then zeta_p:=z; zeta_p2:=1/(b*q)
        else zeta_p:=z; zeta_p2:=0 fi
     else
      if is(z=1) then zeta_m:=z; zeta_m2:=0
       else zeta_m:=z; zeta_m2:=1 fi
    fi;
    for l from 1 to n
     do
       m:=l;
       Q:=pQ(Q)
     od;
   Q
end:

pM:=proc(z,i,j,r,f)
  global R,m,zeta_p,zeta_p2,zeta_m,zeta_m2;
  local l,_R;
   R:=r;
   _R:=(pR@@0)(f);
    if r=‘+‘ then
       if is(z=0) then zeta_p:=z; zeta_p2:=1/(b*q)
        else zeta_p:=z; zeta_p2:=0 fi
     else
      if is(z=1) then zeta_m:=z; zeta_m2:=0
       else zeta_m:=z; zeta_m2:=1 fi
    fi;
    for l from j to i by -1
     do
       m:=l;
       _R:=pR(_R)
     od;
   _R
end:

pN:=proc(z,n,r,f)
   pM(z,1,n,r,f)
end:

pW:=proc(zp,zm,i,j,r,f)
  global R,m,zeta_p,zeta_p2,zeta_m,zeta_m2;
  local l,V;
   R:=r;
   V:=(pF@@0)(f);
   if is(zp=0) then zeta_p:=zp; zeta_p2:=1/(b*q)
     else zeta_p:=zp; zeta_p2:=0 fi;
   if is(zm=0) then zeta_m:=zm; zeta_m2:=1
     else zeta_m:=zm; zeta_m2:=0 fi;
    for l from j to i by -1
     do
       m:=l;
       V:=pF(V)

40 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

     od;
   V
end:

pY:=proc(zp,zm,n,r,f)
   pW(zp,zm,1,n,r,f)
end:

integrate_canonical_form:=proc(n,Q,r)
  global R;
  local Omega,z,z2,i,j,l,p,p2,qq,q2,e,epsilon,epsilon2,
        pom,O,sO,oO,stp,stm,sm,sp,t;
  R:=r;
    if r=‘+‘ then
      z:=0; z2:=1/(b*q)
     else
      z:=1; z2:=0
    fi;
   Omega[z]:=[]; Omega[z2]:=[]; stp:=0; stm:=0;
   p[z]:=0; p[z2]:=0; qq[z]:=0; qq[z2]:=0;
    for i from 0 to n
     do
        if is(Q[i][3]=z) then
          Omega[z]:=[op(Omega[z]),i];
          p[z]:=max(p[z],Q[i][1]);
          qq[z]:=max(qq[z],Q[i][1]+Q[i][2]);
        else
          Omega[z2]:=[op(Omega[z2]),i];
          p[z2]:=max(p[z2],Q[i][1]);
          qq[z2]:=max(qq[z2],Q[i][1]+Q[i][2])
        fi
     od;
   qq[z]:=qq[z]-p[z];
   qq[z2]:=qq[z2]-p[z2];
    if is(p[z]+qq[z]<p[z2]+qq[z2]) then epsilon:=z2; epsilon2:=z
     else epsilon:=z; epsilon2:=z2 fi;
   p2:=max(p[epsilon],p[epsilon2]+qq[epsilon2]);
   q2:=p[epsilon]+qq[epsilon]-p2;
    for e in [epsilon,epsilon2]
     do
      for j in Omega[e]
       do
         pom:=eps()^Q[j][1]*(lambda_pD@@Q[j][4])(poly_to_oper(Q[j][5],b[k]));
         pom:=pU(Q[j][3],Q[j][2],r,pom);
         pom:=pM(e,qq[e]+1,qq[e]+p[e]-Q[j][1],r,
                 pS(e,qq[e]+p[e]-Q[j][1],Q[j][2]+1,r,pom));
         sp:=degree(Q[j][5],x)+Q[j][4]+qq[e]+p[e]-Q[j][1];
         sm:=degree(Q[j][5],x)+Q[j][4]+p[e]-Q[j][1];
          if is(e=epsilon) then
            O[j]:=pM(e,q2+1,qq[e],r,pom);
            sm:=sm+qq[e]-q2
           else
            O[j]:=pZ(epsilon,q2,r,pN(e,p2-p[e],r,
                     pS(e,p2-p[e],qq[e]+1,r,pom)));
            sp:=sp+p2+q2-p[e]-qq[e];
            sm:=sm+p2-p[e]
          fi;
        stp:=max(stp,sp);
        stm:=max(stm,sm)
       od

     od;
   sO:=0;
   oO:=sum(O[t],t=0..n);
    for l from -stm to stp
      do
        sO:=sO+factor(qsimplify(coeff(oO,b[k+l],1)))*b[k+l]
      od;
   [epsilon,q2,p2,sO,stm,stp]
end:

p_i_m:=proc(zeta,_i,_r,_pMp,_stmp,_stpp,eta,_j,_s,_pMm,_stmm,_stpm)
  global R,m,zeta_p,zeta_p2,zeta_m,zeta_m2;
  local p,qq,pom,pUp,pUm,z,sO,oO,stm,stp,l,
        i,r,pMp,stmp,stpp,j,s,pMm,stmm,stpm,op;
   i:=_i;r:=_r;pMp:=_pMp;stmp:=_stmp;stpp:=_stpp;
   j:=_j;s:=_s;pMm:=_pMm;stmm:=_stmm;stpm:=_stpm;
    if is(zeta=0) then zeta_p:=zeta; zeta_p2:=1/(b*q)
      else zeta_p:=zeta; zeta_p2:=0 fi;
    if is(eta=0) then zeta_m:=eta; zeta_m2:=1
      else zeta_m:=eta; zeta_m2:=0 fi;
    if (i>0 and j>0 and (zeta_p-zeta_m2)*(zeta_m-zeta_p2)=0) then
      if i<j then
        r:=r+i; pMp:=pN(zeta,i,‘+‘,pMp); stpm:=stpm+i; i:=0;
       else
        s:=s+j; pMm:=pN(eta,j,‘-‘,pMm); stmm:=stmm+i; j:=0;
      fi
    fi;
   p:=min(i,j); qq:=max(i,j);
    if is(qq=i) then
      op:=[‘+‘,zeta_p,qq-p,p,zeta_p,zeta_m2,s,r];
      R:=‘-‘;
      m:=p;
      pom:=((pDelta)@@r)(pMm);
      pom:=pY(zeta_p2,zeta_m,p,‘-‘,pom);
      z:=zeta_p;
      zeta_p:=zeta_p2;
      zeta_p2:=z;
      R:=‘+‘;
      pUm:=q^(r*s)*pZ(zeta_p,qq-p,‘+‘,pom);
      m:=qq;
      pom:=((pDelta)@@s)(pMp);
      pUp:=pW(zeta_p,zeta_m2,qq-p+1,qq,‘+‘,pom);
      stm:=max(r+p+stmm,s+p+stmp);
      stp:=max(r+qq-p+stpm,s+stpp)
     else
      op:=[‘-‘,zeta_m,qq-p,p,zeta_p,zeta_m2,s,r];
      R:=‘+‘;
      m:=p;
      pom:=((pDelta)@@s)(pMp);
      pom:=pY(zeta_p,zeta_m2,p,‘+‘,pom);
      z:=zeta_m;
      zeta_m:=zeta_m2;
      zeta_m2:=z;
      R:=‘-‘;
      pUp:=pZ(zeta_m,qq-p,‘-‘,pom);
      m:=qq;
      pom:=((pDelta)@@r)(pMm);
      pUm:=q^(r*s)*pW(zeta_p2,zeta_m,qq-p+1,qq,‘-‘,pom);
      stm:=max(r+p+stmm,s+p+stmp);
      stp:=max(s+qq-p+stpp,r+stpm)

42 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

    fi;
   sO:=0;
   oO:=pUp+pUm;
    for l from -stm to stp
      do
        sO:=sO+factor(qsimplify(coeff(oO,b[k+l],1)))*b[k+l]
      od;
   [sO,stm,stp,op]
end:

l_qJacobi:=proc(lm,lp)
  options ‘Copyright 2001 by Pawel Wozny‘;
  global R;
  local Pp,Pm,np,nm,Kp,Km,m,n,l,p,i,j,oper,d,A,B,pom,sp,sm,prze,o;
  nm:=nops(lm); np:=nops(lp);
  R:=‘-‘;
  if nm=0 then Pm:=0 else
      nm:=nm-1; Pm:=sum(simplify(lm[n+1])*(bD@@n)(f(x)),n=0..nm)
    fi;
   R:=‘+‘;
    if np=0 then Pp:=0 else
      np:=np-1; Pp:=sum(simplify(lp[l+1])*(bD@@l)(f(x)),l=0..np)
    fi;
   Km:=canonical_form(nm,Pm,‘-‘);
   Kp:=canonical_form(np,Pp,‘+‘);
   m:=integrate_canonical_form(nm,Km,‘-‘);
   p:=integrate_canonical_form(np,Kp,‘+‘);
   oper:=p_i_m(p[1],p[2],p[3],p[4],p[5],p[6],m[1],m[2],m[3],m[4],m[5],m[6]);
   sm:=0; sp:=0; d:=1;
    for i from -oper[2] to oper[3]
     do
       B[i]:=qsimplify(coeff(oper[1],b[k+i],1)*(h(i)));
        if B[i]<>0 then
          sp:=max(sp,i);
          sm:=min(sm,i)
        fi
     od;
   prze:=-sm-floor((sp-sm)/2);
   o:=oper[4];
    for j from sm+prze to sp+prze
     do
       A[j]:=0;
        if B[j-prze]<>0 then
          A[j]:=factor(qsimplify(shift(B[j-prze],k,prze)));
          print(‘A[‘.j.‘]‘=A[j])
        fi
     od;
    if o[1]=‘+‘ then
      print(’s’=prze,’zeta’=o[2],’eta’=o[6],’d’=o[3],’t’=o[4],
            ’u’=o[7],’v’=o[8]);
      print(T=1/’h’(k+prze)*E^’s’*Z[’zeta’]^‘+(d)‘*Y[’zeta’,’eta’]^‘+(t)‘*
            Z[’zeta’]^‘+(t)‘*(D^‘-‘)^’u’*(D^‘+‘)^’v’);
     else
      print(’s’=prze,’epsilon’=o[2],’zeta’=o[5],’eta’=o[6],’d’=o[3],
            ’t’=o[4],’u’=o[7],’v’=o[8]);
      print(T=1/’h’(k+prze)*E^’s’*Z[’epsilon’]^‘-(d)‘*Y[’zeta’,’eta’]^‘+(t)‘*
            Z[’zeta’]^‘+(t)‘*(D^‘-‘)^’u’*(D^‘+‘)^’v’);
    fi:
   [sm+prze,sp+prze,A]
end:

References

[1] R. Álvarez-Nodarse, J. Arvesú and R. J. Yáñez, On the connection and linearization problem for discrete hypergeometric q-polynomials, J. Math. Anal. Appl. 257, 52–78.

[2] R. Álvarez-Nodarse, N. R. Quintero and A. Ronveaux, On the linearization problem involving Pochhammer symbols and their q-analogues, J. Comput. Appl. Math. 107 (1999), 133-146.

[3] R. Álvarez-Nodarse and A. Ronveaux, Recurrence relations for connection coefficients between q-orthogonal polynomials of discrete variables in the non-uniform lattice x(s) = q^{2s}, J. Phys. A: Math. Gen. 29 (1996), 7165–7175.

[4] R. Álvarez-Nodarse, R. Yáñez and J. S. Dehesa, Modified Clebsch-Gordan-type expansions for products of discrete hypergeometric polynomials, J. Comput. Appl. Math. 89 (1998), 171–197.

[5] I. Area, E. Godoy, A. Ronveaux and A. Zarzo, Inversion problems in the q-Hahn tableau, J. Symbolic Comp.

28 (1999), 767–776.

[6] P. L. Artés, J. S. Dehesa, A. Martinez-Finkelstein and J. Sánchez-Ruiz, Linearization and connection coefficients for hypergeometric-type polynomials, J. Comput. Appl. Math. 99 (1998), 15–26.

[7] R. Askey, Orthogonal Polynomials and Special Functions, Regional Conf. Ser. Appl. Math. 21 (SIAM, Philadelphia, PA, 1975).

[8] T. S. Chihara, An Introduction to Orthogonal Polynomials (Gordon and Breach, New York, 1978). [9] M. Foupouagnigni, A. Ronveaux and W. Koepf, The fourth-order q-difference equation satisfied by the first associated q-classical orthogonal polynomials, J. Comput. Appl. Math. 101 (1999), 231–236.

[10] G. Gasper, Projection formulas for orthogonal polynomials of a discrete variable, J. Math. Anal. Appl. 45 (1974), 176–198.

[11] G. Gasper and M. Rahman, Basic hypergeometric series (Cambridge University Press, Cambridge 1990). [12] E. Godoy, A. Ronveaux, A. Zarzo and I. Area, Minimal recurrence relations for connection coefficients between classical orthogonal polynomials: continuous case, J. Comput. Appl. Math. 84 (1997), 257–275.

[13] W. Hahn, Über Orthogonalpolynome, die q-Differenzengleichungen genügen, Math. Nachr. 2 (1949), 4–34. [14] M. N. Hounkonnou, S. Belmehdi and A. Ronveaux, Linearization of arbitrary product of classical orthogonal polynomials, Appl. Math. 27 (2000), 187–196.

[15] R. Koekoek and R. F. Swarttouw, The Askey scheme of hypergeometric orthogonal polynomials and its q-analogue, Fac. Techn. Math. Informatics, Delft Univ. of Technology, Rep. 98-17, Delft (1998).

[16] T. Koornwinder, Compact quantum groups and q-special functions, in: Representations of Lie groups and quantum groups, eds. V. Baldoni and M. A. Picardello, Pitman Research Notes in Mathematics Series, vol. 311 (Longman, New York 1994), pp. 46–128.

[17] W. Koepf and D. Schmersau, Algorithms for classical orthogonal polynomials, Konrad-Zuse-Zentrum Berlin, Preprint SC 96-23, September 1996.

[18] W. Koepf and D. Schmersau, Representations of orthogonal polynomials, J. Comput. Appl. Math. 90, 57–94. [19] S. Lewanowicz, A new approach to the problem of constructing recurrence relations for the Jacobi coefficients, Zastos. Mat. 21 (1991) 303–326.

[20] S. Lewanowicz, Results on the associated Jacobi and Gegenbauer polynomials, J. Comput. Appl. Math. 49 (1993) 137–143.

[21] S. Lewanowicz, Second-order recurrence relation for the linearization coefficients of the classical orthogonal polynomials, J. Comput. Appl. Math. 69 (1996) 159–170.

[22] S. Lewanowicz, Results on the associated classical orthogonal polynomials, J. Comput. Appl. Math. 65 (1995), 215–231.

[23] S. Lewanowicz, Recurrence relations for the connection coefficients of orthogonal polynomials of a discrete variable, J. Comput. Appl. Math. 76 (1996), 213–229.

[24] S. Lewanowicz, Recurrence relations for the connection coefficients of orthogonal polynomials of a discrete variable on the lattice x(s) = q^{2s}, J. Comput. Appl. Math. 99 (1998), 275–286.

[25] S. Lewanowicz, Recursion formulae for basic hypergeometric functions, in Numerical Analysis in the 20th Century, Vol. I: Approximation Theory, eds. J. Wimp & L. Wuytack, J. Comput. Appl. Math. 121 (2000), 297–312.

[26] S. Lewanowicz, Recurrences for the coefficients of series expansions with respect to classical orthogonal polynomials, Appl. Math. 29 (2002), 97–116.

[27] S. Lewanowicz, Representations for the first associated q-classical orthogonal polynomials, J. Comput. Appl.

Math. 150 (2003), 311-327.

44 STANISŁAW LEWANOWICZ AND PAWEŁ WOŹNY

[28] S. Lewanowicz, Construction of recurrences for the coefficients of series expansions in q-classical orthogonal polynomials, J. Comput. Appl. Math. 153 (2003), 295–309.

[29] S. Lewanowicz, E. Godoy, I. Area, A. Ronveaux and A. Zarzo, Recurrence relations for the coefficients of the Fourier series expansions with respect to q-classical orthogonal polynomials, Numerical Algorithms 23 (2000), 31–50.

[30] Y. L. Luke, The special functions and their approximations, 2 vols. (Academic Press, New York 1969). [31] J. C. Medem, Polinomios ortogonales q-semiclásicos, PhD dissertation, Universidad Politécnica de Madrid, 1996.

[32] J. C. Medem, R. Álvarez-Nodarse and F. Marcellán, On the q-classical polynomials: A distributional study, J. Comput. Appl. Math. 135 (2001), 157–196.

[33] A. F. Nikiforov and V. B. Uvarov, Special Functions of Mathematical Physics (Birkhäuser, Basel, 1988). [34] A. F. Nikiforov, S. K. Suslov and V. B. Uvarov, Classical Orthogonal Polynomials of a Discrete Variable, Springer-Verlag, Berlin, 1991.

[35] A. Ronveaux, Orthogonal polynomials: connection and linearization coefficients, in: M. Alfaro et al. (Eds), Proc. Internat. Workshop Orthogonal Polynomials in Math. Phys., Leganés, 1996, pp. 131–142.

[36] A. Ronveaux, I. Area, E. Godoy and A. Zarzo, Lectures on recursive approach to connection and linearization coefficients between polynomials. Introduction, in Proc. Workshop on Special Functions and Differential Equations, Madras, Jan. 13–24, 1997.

[37] A. Ronveaux, S. Belmehdi, E. Godoy and A. Zarzo, Recurrence relations approach for connection coefficients – Applications to classical discrete orthogonal polynomials, in Symmetries and Integrability of Difference Equations, eds. D. Levi, L. Vinet and P. Winternitz, CRM Proceedings and Lecture Notes Series, vol. 9 (1996), AMS, Providence (RI), pp. 321–337.

[38] A. Ronveaux, S. Houkonnou and S. Belmehdi, Generalized linearization problems, J. Phys. A: Math. Gen.

28 (1995) 4423–4430 [39] A. Ronveaux, A. Zarzo I. Area and E. Godoy, Bernstein bases and Hahn-Eberlein orthogonal polynomials, Int. Transf. Spec. Func. 7 (1998), 87-96.

[40] A. Ronveaux, A. Zarzo I. Area and E. Godoy, Classical orthogonal polynomials: Dependence of parameters, J. Comput. Appl. Math. 121 (2000), 95–112.

[41] A. Ronveaux, A. Zarzo and E. Godoy, Recurrence relation for connection coefficients between two families of orthogonal polynomials, J. Comput. Appl. Math. 62 (1995) 67–73.

[42] J. Sánchez-Ruiz, P. Artés, A. Martínez-Finkelshtein and J. S. Dehesa, General linearization formulae for products of continuous hypergeometric-type polynomials, J. Phys. A: Math. Gen. 32 (1999), 1–22.

[43] J. Sánchez-Ruiz and J. S. Dehesa, Expansions in series of orthogonal hypergeometric polynomials, J. Comput. Appl. Math. 89 (1998), 155–170.

[44] R. Szwarc, Linearization and connection coefficients of orthogonal polynomials, Mh. Math. 113 (1992), 319–329.

[45] R. Szwarc, Connection coefficients of orthogonal polynomials, Canad. Math. Bull. 35 (1992), 548–556. [46] R. Szwarc, Nonnegative linearization and quadratic transformation of Askey-Wilson polynomials, Canad.

Math. Bull. 39 (1996), 241–249.

[47] J. Wimp, Computation with recurrence relations (Pitman, Boston 1984).

[48] P. Woźny, Recurrence relations for the coefficients of expansions in classical orthogonal polynomials of a discrete variable Appl. Math. 30 (2003), 89–107.

[49] A. Zarzo, I. Area, E. Godoy and A. Ronveaux, Results for some inversion problems for classical continuous and discrete orthogonal polynomials, J. Phys. A: Math. Gen. 30 (1997) L35–L40.

E-mail address: {Stanislaw.Lewanowicz,Pawel.Wozny}@ii.uni.wroc.pl

Institute of Computer Science, University of Wrocław, 51-151 Wrocław, Poland